

MOSTEK[®]

MICROCOMPUTER SOFTWARE

Reference Manual

**ANSI BASIC
VERSION 5.3**

MOSTEK ANSI BASIC
Version 5.3
Reference Manual
Copyright 1980 BY MOSTEK CORPORATION

Publication no. MK79708

ANSI BASIC Reference Manual

CONTENTS

INTRODUCTION

CHAPTER 1 General Information About ANSI BASIC

CHAPTER 2 ANSI BASIC Commands and Statements

CHAPTER 3 ANSI BASIC Functions

APPENDIX A New Features in ANSI BASIC, Release 5.3

APPENDIX B ANSI BASIC Disk I/O

APPENDIX C Assembly Language Subroutines

APPENDIX D Converting Programs to ANSI BASIC

APPENDIX E Summary of Error Codes and Error Messages

APPENDIX F Mathematical Functions

APPENDIX G ASCII Character Codes

Introduction

ANSI BASIC is the most extensive implementation of BASIC available for the 8080 and Z80 microprocessors. In its fifth major release (Release 5.3), ANSI BASIC meets the ANSI qualifications for BASIC, as set forth in document BSRX3.60-1978.

MOSTEK ANSI BASIC is an extensive implementation of Microsoft BASIC for the Z80 microprocessor. Its features are comparable to the BASICs found on minicomputers and mainframes. MOSTEK ANSI BASIC is among the fastest microprocessor BASICs available. Designed to operate on MOSTEK Development Systems running on FLP-80DOS V2.1, MDDOS or FLP-80DOS/MDX with 48K bytes or more of memory, BASIC provides a sophisticated software development tool.

ANSI BASIC is implemented as an interpreter and is highly suitable for user interactive processing. Programs and data are stored in a compressed internal format to maximize memory utilization. In a 64K system, 28K of user's program and data storage area are available.

MOSTEK ANSI BASIC is equivalent to Microsoft Extended Disk BASIC with the following exceptions:

1. The default record size for Random files in MOSTEK ANSI BASIC is 124 bytes.

2. The MOSTEK FLP-80DOS dataset specification (see section 1 of the FLP-80DOSV2.1 Operations Manual) is used in filename specifications for the BASIC commands OPEN, KILL, NAME, MERGE, LOAD, SAVE and RUN as illustrated in the following examples:

a) Open the random file NEWFIL on disk unit 1, as number 2.

```
OPEN "R",2,"DK1:NEWFIL"
```

b) Load the program file OLDFIL from disk unit 0.

```
LOAD "OLDFIL"
```

3. The file number used in the OPEN and CLOSE statements is restricted to an integer expression between one and six. MOSTEK ANSI BASIC allows up to 6 files to be open at one time.

There are significant differences between the 5.3 release of ANSI BASIC and the previous releases (release 5.1 and earlier). If you have programs written under a previous release of ANSI BASIC, check Appendix A for new features in 5.3 that may affect execution.

The manual is divided into three large chapters plus a number of appendices. Chapter 1 covers a variety of topics, largely pertaining to information representation when using ANSI BASIC. Chapter 2 contains the syntax and semantics of every command and statement in ANSI BASIC, ordered alphabetically. Chapter 3 describes all of ANSI BASIC's intrinsic functions, also ordered alphabetically. The appendices contain information on compatibility with previous versions; plus lists of error messages, ASCII codes, and math functions; and helpful information on assembly language subroutines and disk I/O.

MOSTEK
Z80 MICROCOMPUTER SYSTEMS
ANSI BASIC SOFTWARE INTERPRETER
FEATURES

- Meets ANSI standard on BASIC (x3.60-1978)
- Direct access to CPU I/O Ports
- Ability to read or write any memory location (PEEK, POKE)
- Arrays with up to 255 dimensions
- Dynamic allocation and de-allocation of arrays
- IF...THEN...ELSE AND IF...GO TO (both if's may be nested)
- Direct (immediate) execution of statements
- Error trapping, with error messages in English
- Four variable types: Integer, string, real and double precision real
- Long variable names significant up to 40 characters
- Full PRINT USING capabilities for formatted output
- Extensive program editing facilities
- Trace facilities
- Can call any number of assembly language subroutines
- Boolean (logical) operations
- Supports up to 6 sequential and random access files on floppy disk
- Variable record length in random access files
- Complete set of file manipulation statements
- Occupies only 23k bytes, not including operating system
- Supports console and line printer I/O
- Allows console output to be redirected to the line printer
- WHILE...WEND structured construct

LANGUAGE COMMANDS SUMMARY

Commands:

AUTO	CLEAR	CONT	DELETE	EDIT
FILES	LIST	LLIST	LOAD	MERGE
NEW	NULL	RENUM	RESET	RUN
SAVE	SYSTEM	TRON	TROFF	WIDTH

Program Statements:

CALL	CHAIN	COMMON	DEFDBL	DEF FN
DEFINT	DEFSNG	DEFSTR	DEFUSR	DIM
END	ERASE	ERROR	FOR...NEXT	GOSUB...RETURN
GOTO	IF...THEN(ELSE)	IF...GOTO	LET	ON ERROR GOTO
ON...GOSUB	ON...GOTO	OPTION BASE	POKE	RANDOMIZE
REM	RESUME	STOP	SWAP	WAIT
WHILE...WEND				

Input/Output Statements:

CLOSE	DATA	FIELD	GET	INPUT
INPUT#	KILL	LINE INPUT	LINE INPUT#	LPRINT
LPRINT USING	LSET	NAME	OPEN	OUT
PRINT	PRINT USING	PRINT#	PRINT# USING	PUT
READ	RESTORE	RESET	RSET	WRITE
WRITE#				

Operators:

=	-	+	*	/
^	\	>	<	<=
>=	<>	MOD	NOT	AND
OR	XOR	IMP	EQU	

Arithmetic Functions:

ABS	ATN	CDBL	CINT	COS
CSNG	EXP	ERR	ERL	FIX
FRE	INT	LOG	RND	SGN
SIN	SQR	TAN	USR	VARPTR

String Functions:

ASC	CHR\$	HEXS	INSTR	LEFT\$
LEN	MIDS	OCTS	RIGHT\$	SPACES\$
SPC\$	STR\$	VAL		

Input/Output Functions:

CVI	CVS	CVD	DSKF	EOF
INP	INPUT\$	LOC	LOF	LPOS
MKD\$	MKI\$	MKSS	PEEK	POS
TAB				

CHAPTER 1
GENERAL INFORMATION ABOUT ANSI BASIC

1.1 INITIALIZATION

To enter ANSI BASIC the user types the word BASIC as a Monitor command in FLP-80DOS V2.1, then the sign-on message is printed on the console as shown below:

\$BASIC

ANSI BASIC V5.3 COPYRIGHT MOSTEK CORP. 1979

The sign-on message is followed by the number of free bytes which represent the amount of space available for BASIC program and string variable storage. The user may now enter BASIC commands or statements. To exit BASIC and return to the FLP-80DOS Monitor the user simply enters the SYSTEM command which reboots the operating system. The system functions performed by the following BASIC statements are of particular interest to the user.

a) RESET

The RESET command should be issued anytime a new diskette is inserted and the user wishes to continue executing BASIC disk I/O statements. This guarantees that the proper sector and track maps are in memory during file operations on the newly inserted diskette. When entering BASIC from the Monitor the RESET command is automatically executed by BASIC.

b) LPRINT AND LLIST

The LPRINT and LLIST statements in BASIC output data to logical unit 5 of the FLP-80DOS operating system. Logical unit 5 is defined during the operating system SYSGEN (System Generation) procedure. Logical unit 5 is typically assigned to the system output listing device (CP:). Prior to execution of BASIC the user may reassign logical unit 5 to a different device (e.g., TT:) using the Monitor ASSIGN command (see section 2 of the FLP-80DOS V2.1 Operations Manual). Version 2.1 or higher of FLP-80DOS is required in ANSI BASIC.

c) POKE 30,1 To redirect console output during program execution to the listing device execute the BASIC command POKE 30,1 . This is convenient in cases such as when the output of PRINT statements is needed in the line printer. To return the console output to logical unit 1, execute POKE 30,0 .

1.2 MODES OF OPERATION

When ANSI BASIC is initialized, it types the prompt "Ok". "Ok" means ANSI BASIC is at command level, that is, it is ready to accept commands. At this point, ANSI BASIC may be used in either of two modes: the direct mode or the indirect mode.

In the direct mode, BASIC statements and commands are not preceded by line numbers. They are executed as they are entered. Results of arithmetic and logical operations may be displayed immediately and stored for later use, but the instructions themselves are lost after execution. This mode is useful for debugging and for using BASIC as a "calculator" for quick computations that do not require a complete program.

The indirect mode is the mode used for entering programs. Program lines are preceded by line numbers and are stored in memory. The program stored in memory is executed by entering the RUN command.

1.3 LINE FORMAT

Program lines in a BASIC program have the following format (square brackets indicate optional):

nnnnn BASIC statement[:BASIC statement...] <carriage return>

At the programmer's option, more than one BASIC statement may be placed on a line, but each statement on a line must be separated from the last by a colon.

A BASIC program line always begins with a line number, ends with a carriage return, and may contain a maximum of 255 characters.

In MOSTEK ANSI BASIC, it is possible to extend a logical line over more than one physical line by use of the terminal's <line feed> key. <Line feed> lets you continue typing a logical line on the next physical line without entering a <carriage return>.

1.3.1 Line Numbers

Every BASIC program line begins with a line number. Line numbers indicate the order in which the program lines are stored in memory and are also used as references when branching and editing. Line numbers must be in the range 0 to 65529. A period (.) may be used in EDIT, LIST, AUTO and DELETE commands to refer to the current line.

1.4 CHARACTER SET

The ANSI BASIC character set is comprised of alphabetic characters, numeric characters and special characters.

The alphabetic characters in ANSI BASIC are the upper case and lower case letters of the alphabet.

The numeric characters in ANSI BASIC are the digits 0 through 9.

The following special characters and terminal keys are recognized by ANSI BASIC:

<u>Character</u>	<u>Name</u>
	Blank
+	Plus sign
-	Minus sign
*	Asterisk or multiplication symbol
/	Slash or division symbol
^	Up arrow or exponentiation symbol
(Left parenthesis
)	Right parenthesis
%	Percent
#	Number (or pound) sign
\$	Dollar sign
!	Exclamation point
[Left bracket
]	Right bracket
,	Comma
.	Period or decimal point
'	Single quotation mark (apostrophe)
;	Semicolon
:	Colon
&	Ampersand
?	Question mark
<	Less than
>	Greater than
\	Backslash or integer division symbol
@	At-sign
_	Underscore
<rubout>	Deletes last character typed.
<escape>	Escapes Edit Mode subcommands. See Section 2.16.
<tab>	Moves print position to next tab stop. Tab stops are every eight columns.
<line feed>	Moves to next physical line.
<carriage return>	Terminates input of a line. See Section 2.16.
bcommands.	Moves print position to next tab stop. Tab stops are every eight columns.
<tab>	Moves print position to next tab stop. Tab stops are every eight columns.
<line feed>	Moves to next physical line.
<carriage return>	Terminates input of a line.

1.4.1 Control Characters

The following control characters are in ANSI BASIC:

- Control-A Enters Edit Mode on the line being typed.
- Control-C Interrupts program execution and returns to ANSI BASIC command level.
- Control-G Rings the bell at the terminal.
- Control-H Backspace. Deletes the last character typed.
- Control-I Tab. Tab stops are every eight columns.
- Control-O Halts program output while execution continues. A second Control-O restarts output.
- Control-R Retypes the line that is currently being typed.
- Control-S Suspends program execution.
- Control-Q Resumes program execution after a Control-S.
- Control-U Deletes the line that is currently being typed.

1.5 CONSTANTS

Constants are the actual values BASIC uses during execution. There are two types of constants: string and numeric. A string constant is a sequence of up to 255 alphanumeric characters enclosed in double quotation marks. Examples of string constants:

- "HELLO"
- "\$25,000.00"
- "Number of Employees"

Numeric constants are positive or negative numbers. Numeric constants in BASIC cannot contain commas. There are five types of numeric constants:

1. Integer constants Whole numbers between -32768 and +32767. Integer constants do not have decimal points.
2. Fixed Point constants Positive or negative real numbers, i.e., numbers that contain decimal points.

3. Floating Point constants

Positive or negative numbers represented in exponential form (similar to scientific notation). A floating point constant consists of an optionally signed integer or fixed point number (the mantissa) followed by the letter E and an optionally signed integer (the exponent). The exponent must be in the range -38 to +38.

Examples:

235.988E-7 = .0000235988
2359E6 = 2359000000

(Double precision floating point constants use the letter D instead of E. See Section 1.5.1.)

4. Hex constants

Hexadecimal numbers with the prefix &H. Examples:

&H76
&H32F

5. Octal constants

Octal numbers with the prefix &O or &. Examples:

&O347
&1234

1.5.1 Single And Double Precision Form For Numeric Constants

Numeric constants may be either single precision or double precision numbers. With double precision, the numbers are stored with 16 digits of precision, and printed with up to 16 digits.

A single precision constant is any numeric constant that has:

1. seven or fewer digits, or
2. exponential form using E, or
3. a trailing exclamation point (!)

A double precision constant is any numeric constant that has:

1. eight or more digits, or
2. exponential form using D, or
3. a trailing number sign (#)

Examples:

<u>Single Precision Constants</u>	<u>Double Precision Constants</u>
46.8	345692811
-1.09E-06	-1.09432D-06
3489.0	3489.0#
22.5!	7654321.1234

1.6 VARIABLES

Variables are names used to represent values that are used in a BASIC program. The value of a variable may be assigned explicitly by the programmer, or it may be assigned as the result of calculations in the program. Before a variable is assigned a value, its value is assumed to be zero.

1.6.1 Variable Names And Declaration Characters

ANSI BASIC variable names may be any length, however, up to 40 characters are significant. The characters allowed in a variable name are letters, numbers and the decimal point. The first character must be a letter. Special type declaration characters are also allowed -- see below.

A variable name may not be a reserved word, however embedded reserved words are allowed. If a variable begins with FN, it is assumed to be a call to a user-defined function. Reserved words include all ANSI BASIC commands, statements, function names and operator names.

Variables may represent either a numeric value or a string. String variable names are written with a dollar sign (\$) as the last character. For example: AS = "SALES REPORT". The dollar sign is a variable type declaration character, that is, it "declares" that the variable will represent a string.

Numeric variable names may declare integer, single or double precision values. The type declaration characters for these variable names are as follows:

```
% Integer variable  
! Single precision variable  
# Double precision variable
```

The default type for a numeric variable name is single precision.

Examples of ANSI BASIC variable names follow.

```
PI#      declares a double precision value  
MINIMUM! declares a single precision value  
LIMIT%    declares an integer value  
N$       declares a string value  
ABC      represents a single precision value
```

There is a second method to declare variable types. ANSI BASIC statements DEFINT, DEFSTR, DEFSNG and DEFDBL may be included in program to declare the types for certain variable names. These statements are described in detail in Section 2.10.

1.6.2 Array Variables

An array is a group or table of values referenced by the same variable name. Each element in an array is referenced by an array variable that is subscripted with integers or integer expressions. An array variable name has as many subscripts as there are dimensions in the array. For example V(10) would reference a value in a one-dimensional array, T(1,4) would reference a value in a two-dimensional array, and so on.

1.7 TYPE CONVERSION

When necessary, BASIC will convert a numeric constant from one type to another. The following rules and examples should be kept in mind.

1. If a numeric constant of one type is set equal to a numeric variable of a different type, the number will be stored as the type declared in the variable name. (If a string variable is set equal to a numeric value or vice versa, a "Type mismatch" error occurs.)

Example:

```
10 A% = 23.42
20 PRINT A%
RUN
23
```

2. During expression evaluation, all of the operands in an arithmetic or relational operation are converted to the same degree of precision, i.e., that of the most precise operand. Also, the result of an arithmetic operation is returned to this degree of precision.

Examples:

```
10 D# = 6#/7      The arithmetic was performed
20 PRINT D#       in double precision and the
RUN              result was returned in D#
                 .8571428571428571 as a double precision value.
```

```
10 D = 6#/7      The arithmetic was performed
20 PRINT D        in double precision and the
RUN              result was returned to D (single
                 .857143      precision variable), rounded and
                           printed as a single precision
                           value.
```

3. Logical operators (see Section 1.8.3) convert their operands to integers and return an integer result. Operands must be in the range -32768 to 32767 or an "Overflow" error occurs.
4. When a floating point value is converted to an integer, the fractional portion is rounded.

Example:

```
10 C% = 55.88
20 PRINT C%
RUN
56
```

5. If a double precision variable is assigned a single precision value, only the first seven digits, rounded, of the converted number will be valid. This is because only seven digits of accuracy were supplied with the single precision value. The absolute value of the difference between the printed double precision number and the original single precision value will be less than 6.3E-8 times the original single precision value.

Example:

```
10 A = 2.04
20 B# = A
30 PRINT A;B#
RUN
2.04 2.039999961853027
```

1.8 EXPRESSIONS AND OPERATORS

An expression may be simply a string or numeric constant, or a variable, or it may combine constants and variables with operators to produce a single value.

Operators perform mathematical or logical operations on values. The operators provided by ANSI BASIC may be divided into four categories:

1. Arithmetic
2. Relational
3. Logical
4. Functional

1.8.1 Arithmetic Operators

The arithmetic operators, in order of precedence, are:

<u>Operator</u>	<u>Operation</u>	<u>Sample Expression</u>
$^$	Exponentiation	X^Y
$-$	Negation	$-X$
$*, /$	Multiplication, Floating Point Division	$X * Y$ X / Y
$+, -$	Addition, Subtraction	$X + Y$

To change the order in which the operations are performed, use parentheses. Operations within parentheses are performed first. Inside parentheses, the usual order of operations is maintained.

Here are some sample algebraic expressions and their BASIC counterparts.

<u>Algebraic Expression</u>	<u>BASIC Expression</u>
$X+2Y$	$X+Y*2$
$X - \frac{Y}{Z}$	$X-Y/Z$
$\frac{XY}{Z}$	$X*Y/Z$
$\frac{X+Y}{Z}$	$(X+Y)/Z$
$(X^2)^Y$	$(X^2)^Y$
$\frac{Z}{Y}$ X	$X^(Y^Z)$
$X(-Y)$	$X*(-Y)$ Two consecutive operators must be separated by parentheses.

1.8.1.1 Integer Division And Modulus Arithmetic -

Two additional operators are available: Integer division and modulus arithmetic.

Integer division is denoted by the backslash (\). The operands are rounded to integers (must be in the range -32768 to 32767) before the division is performed, and the quotient is truncated to an integer. For example:

```
10\4 = 2
25.68\6.99 = 3
```

The precedence of integer division is just after multiplication and floating point division.

Modulus arithmetic is denoted by the operator MOD. It gives the integer value that is the remainder of an integer division. For example:

```
10.4 MOD 4 = 2 (10/4=2 with a remainder 2)
25.68 MOD 6.99 = 5 (26/7=3 with a remainder 5)
```

The precedence of modulus arithmetic is just after integer division.

1.8.1.2 Overflow And Division By Zero -

If, during the evaluation of an expression, a division by zero is encountered, the "Division by zero" error message is displayed, machine infinity with the sign of the numerator is supplied as the result of the division, and execution continues. If the evaluation of an exponentiation results in zero being raised to a negative power, the "Division by zero" error message is displayed, positive machine infinity is supplied as the result of the exponentiation, and execution continues.

If overflow occurs, the "Overflow" error message is displayed, machine infinity with the algebraically correct sign is supplied as the result, and execution continues.

1.8.2 Relational Operators

Relational operators are used to compare two values. The result of the comparison is either "true" (-1) or "false" (0). This result may then be used to make a decision regarding program flow. (See IF, Section 2.25.)

<u>Operator</u>	<u>Relation Tested</u>	<u>Expression</u>
=	Equality	X=Y
<>	Inequality	X<>Y
<	Less than	X<Y
>	Greater than	X>Y
<=	Less than or equal to	X<=Y
>=	Greater than or equal to	X>=Y

(The equal sign is also used to assign a value to a variable. See LET, Section 2.29.)

When arithmetic and relational operators are combined in one expression, the arithmetic is always performed first. For example, the expression

X+Y < (T-1)/Z

is true if the value of X plus Y is less than the value of T-1 divided by Z. More examples:

```
IF SIN(X)<0 GOTO 1000
IF I MOD J <> 0 THEN K=K+1
```

1.8.3 Logical Operators

Logical operators perform tests on multiple relations, bit manipulation, or Boolean operations. The logical operator returns a bitwise result which is either "true" (not zero) or "false" (zero). In an expression, logical operations are performed after arithmetic and relational operations. The outcome of a logical operation is determined as shown in the following table. The operators are listed in order of precedence.

NOT

X	NOT X
1	0
0	1

AND

X	Y	X AND Y
1	1	1
1	0	0
0	1	0
0	0	0

OR

X	Y	X OR Y
1	1	1
1	0	1
0	1	1
0	0	0

XOR

X	Y	X XOR Y
1	1	0
1	0	1
0	1	1
0	0	0

IMP

X	Y	X IMP Y
1	1	1
1	0	0
0	1	1
0	0	1

EQV

X	Y	X EQV Y
1	1	1
1	0	0
0	1	0
0	0	1

Just as the relational operators can be used to make decisions regarding program flow, logical operators can connect two or more relations and return a true or false value to be used in a decision (see IF, Section 2.25). For

example:

```
IF D<200 AND F<4 THEN 80
IF I>10 OR K<0 THEN 50
IF NOT P THEN 100
```

Logical operators work by converting their operands to sixteen bit, signed, two's complement integers in the range -32768 to +32767. (If the operands are not in this range, an error results.) If both operands are supplied as 0 or -1, logical operators return 0 or -1. The given operation is performed on these integers in bitwise fashion, i.e., each bit of the result is determined by the corresponding bits in the two operands.

Thus, it is possible to use logical operators to test bytes for a particular bit pattern. For instance, the AND operator may be used to "mask" all but one of the bits of a status byte at a machine I/O port. The OR operator may be used to "merge" two bytes to create a particular binary value. The following examples will help demonstrate how the logical operators work.

63 AND 16=16	63 = binary 111111 and 16 = binary 10000, so 63 AND 16 = 16
15 AND 14=14	15 = binary 1111 and 14 = binary 1110, so 15 AND 14 = 14 (binary 1110)
-1 AND 8=8	-1 = binary 1111111111111111 and 8 = binary 1000, so -1 AND 8 = 8
4 OR 2=6	4 = binary 100 and 2 = binary 10, so 4 OR 2 = 6 (binary 110)
10 OR 10=10	10 = binary 1010, so 1010 OR 1010 = 1010 (10)
-1 OR -2=-1	-1 = binary 1111111111111111 and -2 = binary 1111111111111110, so -1 OR -2 = -1. The bit complement of sixteen zeros is sixteen ones, which is the two's complement representation of -1.
NOT X=-(X+1)	The two's complement of any integer is the bit complement plus one.

1.8.4 Functional Operators

A function is used in an expression to call a predetermined operation that is to be performed on an operand. ANSI BASIC has "intrinsic" functions that reside in the system, such as SQR (square root) or SIN (sine). All of ANSI BASIC's intrinsic functions are described in Chapter 3. ANSI BASIC also allows "user defined" functions that are written by the programmer. See DEF FN, Section 2.9.

1.8.5 String Operations

Strings may be concatenated using +. For example:

```
10 A$="FILE" : B$="NAME"
20 PRINT A$ + B$
30 PRINT "NEW " + A$ + B$
RUN
FILENAME
NEW FILENAME
```

Strings may be compared using the same relational operators that are used with numbers:

```
=    <>    <    >    <=    >=
```

String comparisons are made by taking one character at a time from each string and comparing the ASCII codes. If all the ASCII codes are the same, the strings are equal. If the ASCII codes differ, the lower code number precedes the higher. If, during string comparison, the end of one string is reached, the shorter string is said to be smaller. Leading and trailing blanks are significant. Examples:

```
"AA" < "AB"
"FILENAME" = "FILENAME"
"X&" > "X#"
"CL " > "CL"
"kg" > "KG"
"SMYTH" < "SMYTHE"
B$ < "9/12/78"      where B$ = "8/12/78"
```

Thus, string comparisons can be used to test string values or to alphabetize strings. All string constants used in comparison expressions must be enclosed in quotation marks.

1.9 INPUT EDITING

If an incorrect character is entered as a line is being typed, it can be deleted with the RUBOUT key or with Control-H. Rubout surrounds the deleted character(s) with backslashes, and Control-H has the effect of backspacing over a character and erasing it. Once a character(s) has been deleted, simply continue typing the line as desired.

To delete a line that is in the process of being typed, type Control-U. A carriage return is executed automatically after the line is deleted.

To correct program lines for a program that is currently in memory, simply retype the line using the same line number. ANSI BASIC will automatically replace the old line with the new line.

More sophisticated editing capabilities are provided in the EDIT command, see Section 2.14.

To delete the entire program that is currently residing in memory, enter the NEW command. (See Section 2.40.) NEW is usually used to clear memory prior to entering a new program.

1.10 ERROR MESSAGES

If ANSI BASIC detects an error that causes program execution to terminate, an error message is printed.

For a complete list of ANSI BASIC error codes and error messages see Appendix E.

CHAPTER 2

ANSI BASIC COMMANDS AND STATEMENTS

All of the ANSI BASIC commands and statements are described in this chapter. Each description is formatted as follows:

Format: Shows the correct format for the instruction.
See below for format notation.

Purpose: Tells what the instruction is used for.

Remarks: Describes in detail how the instruction is used.

Example: Shows sample programs or program segments that demonstrate the use of the instruction.

Format Notation

Wherever the format for a statement or command is given, the following rules apply:

1. Items in capital letters must be input as shown.
2. Items in lower case letters enclosed in angle brackets (< >) are to be supplied by the user.
3. Items in square brackets ([]) are optional.
4. All punctuation except angle brackets and square brackets (i.e., commas, parentheses, semicolons, hyphens, equal signs) must be included where shown.
5. Items followed by an ellipsis (...) may be repeated any number of times (up to the length of the line).

2.1 AUTO

Format: AUTO [<line number>[,<increment>]]

Purpose: To generate a line number automatically after every carriage return.

Remarks: AUTO begins numbering at <line number> and increments each subsequent line number by <increment>. The default for both values is 10. If <line number> is followed by a comma but <increment> is not specified, the last increment specified in an AUTO command is assumed.

If AUTO generates a line number that is already being used, an asterisk is printed after the number to warn the user that any input will replace the existing line. However, typing a carriage return immediately after the asterisk will save the line and generate the next line number.

AUTO is terminated by typing Control-C. The line in which Control-C is typed is not saved. After Control-C is typed, BASIC returns to command level.

Example: AUTO 100,50 Generates line numbers 100,
 150, 200 ...

 AUTO Generates line numbers 10,
 20, 30, 40 ...

2.2 CALL

Format: CALL <variable name>[(<argument list>)]

Purpose: To call an assembly language subroutine.

Remarks: The CALL statement is one way to transfer program flow to an assembly language subroutine. (See also the USR function, Section 3.42) <variable name> contains an address that is the starting point in memory of the subroutine. <variable name> may not be an array variable name. <argument list> contains the arguments that are passed to the assembly language subroutine.

The CALL statement generates the same calling sequence used by Microsoft's FORTRAN, COBOL and BASIC compilers.

Example: 110 MYROUT=&HD000
120 CALL MYROUT(I,J,K)

•

•

•

2.3 CHAIN

Format: CHAIN [MERGE] <filename>[,,[<line number exp>]
 [,ALL][,DELETE<range>]]

Purpose: To call a program and pass variables to it from
the current program.

Remarks: <filename> is the name of the program that is
called. Example:

 CHAIN "PROG1"

<line number exp> is a line number or an expression that evaluates to a line number in the called program. It is the starting point for execution of the called program. If it is omitted, execution begins at the first line.
Example:

 CHAIN "PROG1",1000

<line number exp> is not affected by a RENUM command.

With the ALL option, every variable in the current program is passed to the called program. If the ALL option is omitted, the current program must contain a COMMON statement to list the variables that are passed. See Section 2.7.
Example:

 CHAIN "PROG1",1000,ALL

If the MERGE option is included, it allows a subroutine to be brought into the BASIC program as an overlay. That is, a MERGE operation is performed with the current program and the called program. The called program must be an ASCII file if it is to be MERGED. Example:

 CHAIN MERGE "OVRLAY",1000

After an overlay is brought in, it is usually desirable to delete it so that a new overlay may be brought in. To do this, use the DELETE option. Example:

 CHAIN MERGE "OVRLAY2",1000,DELETE 1000-5000

The line numbers in <range> are affected by the RENUM command.

NOTE: The Microsoft BASIC compiler does not support the ALL, MERGE, and DELETE options to CHAIN. If you wish to maintain compatibility with the BASIC compiler, it is recommended that COMMON be used to pass variables and that overlays not be used.

2.4 CLEAR

Format: CLEAR [,,[<expression1>][,<expression2>]]

Purpose: To set all numeric variables to zero and all string variables to null; and, optionally, to set the end of memory and the amount of stack space.

Remarks: <expression1> is a memory location which, if specified, sets the highest location available for use by ANSI BASIC.

<expression2> sets aside stack space for BASIC. The default is 1000 bytes or one-eighth of the available memory, whichever is smaller.

NOTE: In previous versions of ANSI BASIC, <expression1> set the amount of string space, and <expression2> set the end of memory. ANSI BASIC, release 5.3 and later, allocates string space dynamically. An "Out of string space error" occurs only if there is no free memory left for BASIC to use.

Examples: CLEAR

CLEAR ,32768

CLEAR ,,2000

CLEAR,32768,2000

2.5 CLOSE

Format: CLOSE[[#]<file number>[,[#]<file number...>]]

Purpose: To conclude I/O to a disk file.

Remarks: <file number> is the number under which the file was OPENed. A CLOSE with no arguments closes all open files.

The association between a particular file and file number terminates upon execution of a CLOSE. The file may then be reOPENed using the same or a different file number; likewise, that file number may now be reused to OPEN any file.

A CLOSE for a sequential output file writes the final buffer of output.

The END statement and the NEW command always CLOSE all disk files automatically. (STOP does not close disk files.)

Example: See Appendix B.

2.6 COMMON

Format: COMMON <list of variables>

Purpose: To pass variables to a CHAINed program.

Remarks: The COMMON statement is used in conjunction with the CHAIN statement. COMMON statements may appear anywhere in a program, though it is recommended that they appear at the beginning. The same variable cannot appear in more than one COMMON statement. Array variables are specified by appending "()" to the variable name. If all variables are to be passed, use CHAIN with the ALL option and omit the COMMON statement.

Example: 100 COMMON A,B,C,D(),GS
110 CHAIN "PROG3",10
•
•
•

2.7 CONT

Format: CONT

Purpose: To continue program execution after a Control-C has been typed, or a STOP or END statement has been executed.

Remarks: Execution resumes at the point where the break occurred. If the break occurred after a prompt from an INPUT statement, execution continues with the reprinting of the prompt (?) or prompt string).

CONT is usually used in conjunction with STOP for debugging. When execution is stopped, intermediate values may be examined and changed using direct mode statements. Execution may be resumed with CONT or a direct mode GOTO, which resumes execution at a specified line number.

CONT is invalid if the program has been edited during the break.

Example: See example Section 2.61, STOP.

2.8 DATA

Format: DATA <list of constants>

Purpose: To store the numeric and string constants that are accessed by the program's READ statement(s). (See READ, Section 2.53)

Remarks: DATA statements are nonexecutable and may be placed anywhere in the program. A DATA statement may contain as many constants as will fit on a line (separated by commas), and any number of DATA statements may be used in a program. The READ statements access the DATA statements in order (by line number) and the data contained therein may be thought of as one continuous list of items, regardless of how many items are on a line or where the lines are placed in the program.

<list of constants> may contain numeric constants in any format, i.e., fixed point, floating point or integer. (No numeric expressions are allowed in the list.) String constants in DATA statements must be surrounded by double quotation marks only if they contain commas, colons or significant leading or trailing spaces. Otherwise, quotation marks are not needed.

The variable type (numeric or string) given in the READ statement must agree with the corresponding constant in the DATA statement.

DATA statements may be reread from the beginning by use of the RESTORE statement (Section 2.57).

Example: See examples in Section 2.53, READ.

2.9 DEF FN

Format: DEF FN<name>[(<parameter list>)]=<function definition>

Purpose: To define and name a function that is written by the user.

Remarks: <name> must be a legal variable name. This name, preceded by FN, becomes the name of the function. <parameter list> is comprised of those variable names in the function definition that are to be replaced when the function is called. The items in the list are separated by commas. <function definition> is an expression that performs the operation of the function. It is limited to one line. Variable names that appear in this expression serve only to define the function; they do not affect program variables that have the same name. A variable name used in a function definition may or may not appear in the parameter list. If it does, the value of the parameter is supplied when the function is called. Otherwise, the current value of the variable is used.

The variables in the parameter list represent, on a one-to-one basis, the argument variables or values that will be given in the function call. User-defined functions may be numeric or string. If a type is specified in the function name, the value of the expression is forced to that type before it is returned to the calling statement. If a type is specified in the function name and the argument type does not match, a "Type mismatch" error occurs.

A DEF FN statement must be executed before the function it defines may be called. If a function is called before it has been defined, an "Undefined user function" error occurs. DEF FN is illegal in the direct mode.

Example:

```
410 DEF FNAB(X,Y)=X^3/Y^2  
420 T=FNAB(I,J)
```

Line 410 defines the function FNAB. The function is called in line 420.

2.10 DEFINT/SNG/DBL/STR

Format: DEF<type> <range of letters>
where <type> is INT, SNG, DBL, or STR

Purpose: To declare variable types as integer, single precision, double precision, or string.

Remarks: A DEFtype statement declares that the variable names beginning with the letter(s) specified will be that type variable. However, a type declaration character always takes precedence over a DEFtype statement in the typing of a variable.

If no type declaration statements are encountered, ANSI BASIC assumes all variables without declaration characters are single precision variables.

Examples: 10 DEFDBL L-P All variables beginning with the letters L, M, N, O, and P will be double precision variables.

10 DEFSTR A All variables beginning with the letter A will be string variables.

2.11 DEF USR

Format: DEF USR[<digit>]=<integer expression>

Purpose: To specify the starting address of an assembly language subroutine.

Remarks: <digit> may be any digit from 0 to 9. The digit corresponds to the number of the USR routine whose address is being specified. If <digit> is omitted, DEF USR0 is assumed. The value of <integer expression> is the starting address of the USR routine. See Appendix C, Assembly Language Subroutines.

Any number of DEF USR statements may appear in a program to redefine subroutine starting addresses, thus allowing access to as many subroutines as necessary.

Example:

```
.  
. .  
200 DEF USR0=24000  
210 X=USR0(Y^2/2.89)  
. .  
. .
```

2.12 DELETE

Format: **DELETE**[<line number>][-<line number>]

Purpose: To delete program lines.

Remarks: ANSI BASIC always returns to command level after a DELETE is executed. If <line number> does not exist, an "Illegal function call" error occurs.

Examples: DELETE 40 Deletes line 40

DELETE 40-100 Deletes lines 40 through 100, inclusive

DELETE-40 Deletes all lines up to
and including line 40

2.13 DIM

Format: DIM <list of subscripted variables>

Purpose: To specify the maximum values for array variable subscripts and allocate storage accordingly.

Remarks: If an array variable name is used without a DIM statement, the maximum value of its subscript(s) is assumed to be 10. If a subscript is used that is greater than the maximum specified, a "Subscript out of range" error occurs. The minimum value for a subscript is always 0, unless otherwise specified with the OPTION BASE statement (see Section 2.45).

The DIM statement sets all the elements of the specified arrays to an initial value of zero.

Example:

```
10 DIM A(20)
20 FOR I=0 TO 20
30 READ A(I)
40 NEXT I
      .
      .
      .
```

2.14 EDIT

Format: EDIT <line number>

Purpose: To enter Edit Mode at the specified line.

Remarks: In Edit Mode, it is possible to edit portions of a line without retyping the entire line. Upon entering Edit Mode, ANSI BASIC types the line number of the line to be edited, then it types a space and waits for an Edit Mode subcommand.

Edit Mode Subcommands

Edit Mode subcommands are used to move the cursor or to insert, delete, replace, or search for text within a line. The subcommands are not echoed. Most of the Edit Mode subcommands may be preceded by an integer which causes the command to be executed that number of times. When a preceding integer is not specified, it is assumed to be 1.

Edit Mode subcommands may be categorized according to the following functions:

1. Moving the cursor
2. Inserting text
3. Deleting text
4. Finding text
5. Replacing text
6. Ending and restarting Edit Mode

NOTE

In the descriptions that follow, <ch> represents any character, <text> represents a string of characters of arbitrary length, [i] represents an optional integer (the default is 1), and \$ represents the Escape (or Altmode) key.

1. Moving the Cursor

- Space Use the space bar to move the cursor to the right. [i]Space moves the cursor i spaces to the right. Characters are printed as you space over them.
- Rubout In Edit Mode, [i]Rubout moves the cursor i spaces to the left (backspaces). Characters are printed as you backspace over them.

2. Inserting Text

- I I<text>\$ inserts <text> at the current cursor position. The inserted characters are printed on the terminal. To terminate insertion, type Escape. If Carriage Return is typed during an Insert command, the effect is the same as typing Escape and then Carriage Return. During an Insert command, the Rubout or Delete key on the terminal may be used to delete characters to the left of the cursor. If an attempt is made to insert a character that will make the line longer than 255 characters, a bell (Control-G) is typed and the character is not printed.
- X The X subcommand is used to extend the line. X moves the cursor to the end of the line, goes into insert mode, and allows insertion of text as if an Insert command had been given. When you are finished extending the line, type Escape or Carriage Return.

3. Deleting Text

- D [i]D deletes i characters to the right of the cursor. The deleted characters are echoed between backslashes, and the cursor is positioned to the right of the last character deleted. If there are fewer than i characters to the right of the cursor, iD deletes the remainder of the line.
- H H deletes all characters to the right of the cursor and then automatically enters insert mode. H is useful for replacing statements at the end of a line.

4. Finding Text

- S The subcommand [i]S<ch> searches for the ith occurrence of <ch> and positions the cursor before it. The character at the current cursor position is not included in the search. If <ch> is not found, the cursor will stop at the end of

the line. All characters passed over during the search are printed.

K The subcommand [i]K<ch> is similar to [i]S<ch>, except all the characters passed over in the search are deleted. The cursor is positioned before <ch>, and the deleted characters are enclosed in backslashes.

5. Replacing Text

C The subcommand C<ch> changes the next character to <ch>. If you wish to change the next i characters, use the subcommand iC, followed by i characters. After the ith new character is typed, change mode is exited and you will return to Edit Mode.

6. Ending and Restarting Edit Mode

<cr> Typing Carriage Return prints the remainder of the line, saves the changes you made and exits Edit Mode.

E The E subcommand has the same effect as Carriage Return, except the remainder of the line is not printed.

Q The Q subcommand returns to ANSI BASIC command level, without saving any of the changes that were made to the line during Edit Mode.

L The L subcommand lists the remainder of the line (saving any changes made so far) and repositions the cursor at the beginning of the line, still in Edit Mode. L is usually used to list the line when you first enter Edit Mode.

A The A subcommand lets you begin editing a line over again. It restores the original line and repositions the cursor at the beginning.

NOTE

If ANSI BASIC receives an unrecognizable command or illegal character while in Edit Mode, it prints a bell (Control-G) and the command or character is ignored.

Syntax Errors

When a Syntax Error is encountered during execution of a program, ANSI BASIC automatically enters Edit Mode at the line that caused the error. For example:

```
10 K = 2(4)
RUN
?Syntax error in 10
10
```

When you finish editing the line and type Carriage Return or the E subcommand, ANSI BASIC reinserts the line, which causes all variable values to be lost. To preserve the variable values for examination, first exit Edit Mode with the Q subcommand. ANSI BASIC will return to command level, and all variable values will be preserved.

Control-A

To enter Edit Mode on the line you are currently typing, type Control-A. ANSI BASIC responds with a carriage return, an exclamation point (!) and a space. The cursor will be positioned at the first character in the line. Proceed by typing an Edit Mode subcommand.

NOTE

Remember, if you have just entered a line and wish to go back and edit it, the command "EDIT." will enter Edit Mode at the current line. (The line number symbol "." always refers to the current line.)

2.15 END

Format: END

Purpose: To terminate program execution, close all files and return to command level.

Remarks: END statements may be placed anywhere in the program to terminate execution. Unlike the STOP statement, END does not cause a BREAK message to be printed. An END statement at the end of a program is optional. ANSI BASIC always returns to command level after an END is executed.

Example: 520 IF K>1000 THEN END ELSE GOTO 20

2.16 ERASE

Format: ERASE <list of array variables>

Purpose: To eliminate arrays from a program.

Remarks: Arrays may be redimensioned after they are ERASEd, or the previously allocated array space in memory may be used for other purposes. If an attempt is made to redimension an array without first ERASEing it, a "Redimensioned array" error occurs.

NOTE: The Microsoft BASIC compiler does not support ERASE.

Example:

```
.  
. .  
450 ERASE A,B  
460 DIM B(99)  
. .  
. .
```

2.17 ERR AND ERL VARIABLES

When an error handling subroutine is entered, the variable ERR contains the error code for the error, and the variable ERL contains the line number of the line in which the error was detected. The ERR and ERL variables are usually used in IF...THEN statements to direct program flow in the error trap routine.

If the statement that caused the error was a direct mode statement, ERL will contain 65535. To test if an error occurred in a direct statement, use IF 65535 = ERL THEN ... Otherwise, use

```
IF ERR = error code THEN ...
```

```
IF ERL = line number THEN ...
```

If the line number is not on the right side of the relational operator, it cannot be renumbered by RENUM. Because ERL and ERR are reserved variables, neither may appear to the left of the equal sign in a LET (assignment) statement. ANSI BASIC error codes are listed in Appendix E.

2.18 ERROR

Format: ERROR <integer expression>

Purpose: 1) To simulate the occurrence of an ANSI BASIC error; or 2) to allow error codes to be defined by the user.

Remarks: The value of <integer expression> must be greater than 0 and less than 255. If the value of <integer expression> equals an error code already in use by ANSI BASIC (see Appendix E) the ERROR statement will simulate the occurrence of that error, and the corresponding error message will be printed. (See Example 1.)

To define your own error code, use a value that is greater than any used by ANSI BASIC's error codes. (It is preferable to use the highest available values, so compatibility may be maintained when more error codes are added to ANSI BASIC.) This user-defined error code may then be conveniently handled in an error trap routine. (See Example 2.)

If an ERROR statement specifies a code for which no error message has been defined, ANSI BASIC responds with the message UNPRINTABLE ERROR. Execution of an ERROR statement for which there is no error trap routine causes an error message to be printed and execution to halt.

Example 1: LIST
10 S = 10
20 T = 5
30 ERROR S + T
40 END
Ok
RUN
String too long in line 30

Or, in direct mode:

Ok
ERROR 15 (you type this line)
String too long (ANSI BASIC types this line)
Ok

Example 2:

```
.  
. .  
110 ON ERROR GOTO 400  
120 INPUT "WHAT IS YOUR BET";B  
130 IF B > 5000 THEN ERROR 210  
. .  
. .  
400 IF ERR = 210 THEN PRINT "HOUSE LIMIT IS $5000"  
410 IF ERL = 130 THEN RESUME 120  
. .  
. .
```

2.19 FIELD

Format: FIELD[#]<file number>,<field width> AS <string variable>...

Purpose: To allocate space for variables in a random file buffer.

Remarks: To get data out of a random buffer after a GET or to enter data before a PUT, a FIELD statement must have been executed.

<file number> is the number under which the file was OPENed. <field width> is the number of characters to be allocated to <string variable>. For example,

FIELD 1, 20 AS NS\$, 10 AS ID\$, 40 AS ADD\$

allocates the first 20 positions (bytes) in the random file buffer to the string variable NS\$, the next 10 positions to ID\$, and the next 40 positions to ADD\$. FIELD does NOT place any data in the random file buffer. (See LSET/RSET and GET.)

The total number of bytes allocated in a FIELD statement must not exceed the record length that was specified when the file was OPENed. Otherwise, a "Field overflow" error occurs. (The default record length is 124.)

Any number of FIELD statements may be executed for the same file, and all FIELD statements that have been executed are in effect at the same time.

Example: See Appendix B.

NOTE: Do not use a FIELDed variable name in an INPUT or LET statement. Once a variable name is FIELDed, it points to the correct place in the random file buffer. If a subsequent INPUT or LET statement with that variable name is executed, the variable's pointer is moved to string space.

2.20 FILES

Format: FILES <disk unit number>

Purpose: To print on the console the names of files residing on the specified disk unit.

Remarks: <disk unit number> is a digit ranging from 0-3 depending on the number of disk units sysgened into the system. The file name, extension and user number of each dataset in that disk unit are printed on the console.

Example:

```
FILES
STRTRK.BAS[1]      PIP    .BIN[1]      BASIC.BIN[1]
CHASE .BAS[1]       ELIZA .BAS[1]
OK

FILES 1
EDIT  .BIN[1]       HANG  .BAS[1]      LANDER.BAS[1]
PIP   .BIN[1]       ASM   .BIN[1]       TEST  .BAS[1]
OK
```

2.21 FOR...NEXT

Format: FOR <variable>=x TO y [STEP z]
 .
 .
 .
 NEXT [<variable>][,<variable>...]
 where x, y and z are numeric expressions.

Purpose: To allow a series of instructions to be performed in a loop a given number of times.

Remarks: <variable> is used as a counter. The first numeric expression (x) is the initial value of the counter. The second numeric expression (y) is the final value of the counter. The program lines following the FOR statement are executed until the NEXT statement is encountered. Then the counter is incremented by the amount specified by STEP. A check is performed to see if the value of the counter is now greater than the final value (y). If it is not greater, ANSI BASIC branches back to the statement after the FOR statement and the process is repeated. If it is greater, execution continues with the statement following the NEXT statement. This is a FOR...NEXT loop. If STEP is not specified, the increment is assumed to be one. If STEP is negative, the final value of the counter is set to be less than the initial value. The counter is decremented each time through the loop, and the loop is executed until the counter is less than the final value.

The body of the loop is skipped if the initial value of the loop times the sign of the step exceeds the final value times the sign of the step.

Nested Loops

FOR...NEXT loops may be nested, that is, a FOR...NEXT loop may be placed within the context of another FOR...NEXT loop. When loops are nested, each loop must have a unique variable name as its counter. The NEXT statement for the inside loop must appear before that for the outside loop. If nested loops have the same end point, a single NEXT statement may be used for all of them.

The variable(s) in the NEXT statement may be omitted, in which case the NEXT statement will match the most recent FOR statement. If a NEXT statement is encountered before its corresponding FOR statement, a "NEXT without FOR" error message is issued and execution is terminated.

Example 1: 10 K=10
20 FOR I=1 TO K STEP 2
30 PRINT I;
40 K=K+10
50 PRINT K
60 NEXT
RUN
1 20
3 30
5 40
7 50
9 60
Ok

Example 2: 10 J=0
20 FOR I=1 TO J
30 PRINT I
40 NEXT I

In this example, the loop does not execute because the initial value of the loop exceeds the final value.

Example 3: 10 I=5
20 FOR I=1 TO I+5
30 PRINT I;
40 NEXT
RUN
1 2 3 4 5 6 7 8 9 10
Ok

In this example, the loop executes ten times. The final value for the loop variable is always set before the initial value is set. (Note: Previous versions of ANSI BASIC set the initial value of the loop variable before setting the final value; i.e., the above loop would have executed six times.)

2.22 GET

Format: GET [#]<file number>[,<record number>]

Purpose: To read a record from a random disk file into a random buffer.

Remarks: <file number> is the number under which the file was OPENed. If <record number> is omitted, the next record (after the last GET) is read into the buffer. The largest possible record number is 32767.

Example: See Appendix B.

2.23 GOSUB...RETURN

Format: GOSUB <line number>

.

.

.

RETURN

Purpose: To branch to and return from a subroutine.

Remarks: <line number> is the first line of the subroutine.

A subroutine may be called any number of times in a program, and a subroutine may be called from within another subroutine. Such nesting of subroutines is limited only by available memory.

The RETURN statement(s) in a subroutine cause ANSI BASIC to branch back to the statement following the most recent GOSUB statement. A subroutine may contain more than one RETURN statement, should logic dictate a return at different points in the subroutine. Subroutines may appear anywhere in the program, but it is recommended that the subroutine be readily distinguishable from the main program. To prevent inadvertant entry into the subroutine, it may be preceded by a STOP, END, or GOTO statement that directs program control around the subroutine.

Example:

```
10 GOSUB 40
20 PRINT "BACK FROM SUBROUTINE"
30 END
40 PRINT "SUBROUTINE";
50 PRINT " IN";
60 PRINT " PROGRESS"
70 RETURN
RUN
SUBROUTINE IN PROGRESS
BACK FROM SUBROUTINE
Ok
```

2.24 GOTO

Format: GOTO <line number>

Purpose: To branch unconditionally out of the normal program sequence to a specified line number.

Remarks: If <line number> is an executable statement, that statement and those following are executed. If it is a nonexecutable statement, execution proceeds at the first executable statement encountered after <line number>.

Example:

```
LIST
10 READ R
20 PRINT "R =";R,
30 A = 3.14*R^2
40 PRINT "AREA =";A
50 GOTO 10
60 DATA 5,7,12
Ok
RUN
R = 5           AREA = 78.5
R = 7           AREA = 153.86
R = 12          AREA = 452.16
?Out of data in 10
Ok
```

2.25 IF...THEN[...ELSE] AND IF...GOTO

Format: IF <expression> [,]THEN <statement(s)> ° <line number>
 [ELSE <statement(s)> ° <line number>]

Format: IF <expression> GOTO <line number>
 [ELSE <statement(s)> ° <line number>]

Purpose: To make a decision regarding program flow based
 on the result returned by an expression.

Remarks: If the result of <expression> is not zero, the
 THEN or GOTO clause is executed. THEN may be
 followed by either a line number for branching
 or one or more statements to be executed. GOTO
 is always followed by a line number. If the
 result of <expression> is zero, the THEN or GOTO
 clause is ignored and the ELSE clause, if
 present, is executed. Execution continues with
 the next executable statement.

Nesting of IF Statements

IF...THEN...ELSE statements may be nested.
Nesting is limited only by the length of the
line. For example

IF X>Y THEN PRINT "GREATER" ELSE IF Y>X
 THEN PRINT "LESS THAN" ELSE PRINT "EQUAL"
is a legal statement. If the statement does not
contain the same number of ELSE and THEN
clauses, each ELSE is matched with the closest
unmatched THEN. For example

IF A=B THEN IF B=C THEN PRINT "A=C"
 ELSE PRINT "A<>C"

will not print "A<>C" when A<>B.

If an IF...THEN statement is followed by a line
number in the direct mode, an "Undefined line"
error results unless a statement with the
specified line number had previously been
entered in the indirect mode.

NOTE: When using IF to test equality for a value that is the result of a floating point computation, remember that the internal representation of the value may not be exact. Therefore, the test should be against the range over which the accuracy of the value may vary. For example, to test a computed variable A against the value 1.0, use:

```
IF ABS (A-1.0)<1.0E-6 THEN ...
```

This test returns true if the value of A is 1.0 with a relative error of less than 1.0E-6.

Example 1: 200 IF I THEN GET#1,I

This statement GETs record number I if I is not zero.

Example 2: 100 IF(I<20)*(I>10) THEN DB=1979-1:GOTO 300
110 PRINT "OUT OF RANGE"

In this example, a test determines if I is greater than 10 and less than 20. If I is in this range, DB is calculated and execution branches to line 300. If I is not in this range, execution continues with line 110.

Example 3: 210 IF IOFLAG THEN PRINT A\$ ELSE LPRINT A\$

This statement causes printed output to go either to the terminal or the line printer, depending on the value of a variable (IOFLAG). If IOFLAG is zero, output goes to the line printer, otherwise output goes to the terminal.

2.26 INPUT

Format: INPUT[;][<"prompt string">];]<list of variables>

Purpose: To allow input from the terminal during program execution.

Remarks: When an INPUT statement is encountered, program execution pauses and a question mark is printed to indicate the program is waiting for data. If <"prompt string"> is included, the string is printed before the question mark. The required data is then entered at the terminal.

If INPUT is immediately followed by a semicolon, then the carriage return typed by the user to input data does not echo a carriage return/line feed sequence.

The data that is entered is assigned to the variable(s) given in <variable list>. The number of data items supplied must be the same as the number of variables in the list. Data items are separated by commas.

The variable names in the list may be numeric or string variable names (including subscripted variables). The type of each data item that is input must agree with the type specified by the variable name. (Strings input to an INPUT statement need not be surrounded by quotation marks.)

Responding to INPUT with too many or too few items, or with the wrong type of value (numeric instead of string, etc.) causes the message "?Redo from start" to be printed. No assignment of input values is made until an acceptable response is given.

If there is only one variable in the INPUT list then responding with only a carriage return causes the variable to be assigned the value of zero if its type is numeric, or a blank if its type is string.

Examples:

```
10 INPUT X
20 PRINT X "SQUARED IS" X^2
30 END
RUN
? 5      (The 5 was typed in by the user
           in response to the question mark.)
5 SQUARED IS 25
Ok

LIST
10 PI=3.14
20 INPUT "WHAT IS THE RADIUS";R
30 A=PI*R^2
40 PRINT "THE AREA OF THE CIRCLE IS";A
50 PRINT
60 GOTO 20
Ok
RUN
WHAT IS THE RADIUS? 7.4 (User types 7.4)
THE AREA OF THE CIRCLE IS 171.946

WHAT IS THE RADIUS?
etc.
```

2.27 INPUT#

Format: INPUT#<file number>,<variable list>

Purpose: To read data items from a sequential disk file and assign them to program variables.

Remarks: <file number> is the number used when the file was OPENed for input. <variable list> contains the variable names that will be assigned to the items in the file. (The variable type must match the type specified by the variable name.) With INPUT#, no question mark is printed, as with INPUT.

The data items in the file should appear just as they would if data were being typed in response to an INPUT statement. With numeric values, leading spaces, carriage returns and line feeds are ignored. The first character encountered that is not a space, carriage return or line feed is assumed to be the start of a number. The number terminates on a space, carriage return, line feed or comma.

If ANSI BASIC is scanning the sequential data file for a string item, leading spaces, carriage returns and line feeds are also ignored. The first character encountered that is not a space, carriage return, or line feed is assumed to be the start of a string item. If this first character is a quotation mark ("), the string item will consist of all characters read between the first quotation mark and the second. Thus, a quoted string may not contain a quotation mark as a character. If the first character of the string is not a quotation mark, the string is an unquoted string, and will terminate on a comma, carriage or line feed (or after 255 characters have been read). If end of file is reached when a numeric or string item is being INPUT, the item is terminated.

Example: See Appendix B.

2.28 KILL

Format: KILL <filename>

Purpose: To delete a file from disk.

Remarks: If a KILL statement is given for a file that is currently OPEN, a "File already open" error occurs.

KILL is used for all types of disk files: program files, random data files and sequential data files.

Example: 200 KILL "DATA1"

See also Appendix B.

2.29 LET

Format: [LET] <variable>=<expression>

Purpose: To assign the value of an expression to a variable.

Remarks: Notice the word LET is optional, i.e., the equal sign is sufficient when assigning an expression to a variable name.

Example:

```
110 LET D=12
120 LET E=12^2
130 LET F=12^4
140 LET SUM=D+E+F
```

•
•
•
or

```
110 D=12
120 E=12^2
130 F=12^4
140 SUM=D+E+F
      •
      •
      •
```

2.30 LINE INPUT

Format: LINE INPUT[;][<"prompt string">];]<string variable>

Purpose: To input an entire line (up to 254 characters) to a string variable, without the use of delimiters.

Remarks: The prompt string is a string literal that is printed at the terminal before input is accepted. A question mark is not printed unless it is part of the prompt string. All input from the end of the prompt to the carriage return is assigned to <string variable>.

If LINE INPUT is immediately followed by a semicolon, then the carriage return typed by the user to end the input line does not echo a carriage return/line feed sequence at the terminal.

A LINE INPUT may be escaped by typing Control-C. ANSI BASIC will return to command level and type Ok. Typing CONT resumes execution at the LINE INPUT.

Example: See Example, Section 2.31, LINE INPUT#.

2.31 LINE INPUT#

Format: LINE INPUT#<file number>,<string variable>

Purpose: To read an entire line (up to 254 characters), without delimiters, from a sequential disk data file to a string variable.

Remarks: <file number> is the number under which the file was OPENed. <string variable> is the variable name to which the line will be assigned. LINE INPUT# reads all characters in the sequential file up to a carriage return. It then skips over the carriage return/line feed sequence, and the next LINE INPUT# reads all characters up to the next carriage return. (If a line feed/carriage return sequence is encountered, it is preserved.)

LINE INPUT# is especially useful if each line of a data file has been broken into fields, or if a ANSI BASIC program saved in ASCII mode is being read as data by another program.

Example:

```
10 OPEN "O",1,"LIST"
20 LINE INPUT "CUSTOMER INFORMATION? ";CS
30 PRINT #1, CS
40 CLOSE 1
50 OPEN "I",1,"LIST"
60 LINE INPUT #1, CS
70 PRINT CS
80 CLOSE 1
RUN
CUSTOMER INFORMATION? LINDA JONES      234,4      MEMPHIS
LINDA JONES      234,4      MEMPHIS
Ok
```

2.32 LIST

Format 1: LIST [<line number>]

Format 2: LIST [<line number>[-[<line number>]]]

Purpose: To list all or part of the program currently in memory at the terminal.

Remarks: ANSI BASIC always returns to command level after a LIST is executed.

Format 1: If <line number> is omitted, the program is listed beginning at the lowest line number. (Listing is terminated either by the end of the program or by typing Control-C and interrupted by Control-S.) If line number is included then only the specified line will be listed.

Format 2: This format allows the following options:

1. If only the first number is specified, that line and all higher-numbered lines are listed.
2. If only the second number is specified, all lines from the beginning of the program through that line are listed.
3. If both numbers are specified, the entire range is listed.

Examples:**Format 1:**

LIST Lists the program currently
 in memory.

LIST 500 Lists line 500.

Format 2:

LIST 150- Lists all lines from 150
 to the end.

LIST -1000 Lists all lines from the
 lowest number through 1000.

LIST 150-1000 Lists lines 150 through
 1000, inclusive.

2.33 LLIST

Format: LLIST [<line number>[-[<line number>]]]

Purpose: To list all or part of the program currently in memory at the line printer.

Remarks: LLIST assumes a 132-character wide printer.

ANSI BASIC always returns to command level after an LLIST is executed. The options for LLIST are the same as for LIST, Format 2.

Example: See the examples for LIST, Format 2.

2.43 LOAD

Format: LOAD <filename>[,R]

Purpose: To load a file from disk into memory.

Remarks: <filename> is the name that was used when the file was SAVED.

LOAD closes all open files and deletes all variables and program lines currently residing in memory before it loads the designated program. However, if the "R" option is used with LOAD, the program is RUN after it is LOADED, and all open data files are kept open. Thus, LOAD with the "R" option may be used to chain several programs (or segments of the same program). Information may be passed between the programs using their disk data files.

Example: LOAD "STRTRK",R

2.35 LPRINT AND LPRINT USING

Format: LPRINT [<list of expressions>]
 LPRINT USING <"format string">;<list of expressions>

Purpose: To print data at the line printer.

Remarks: Same as PRINT and PRINT USING, except output goes to the line printer. See Section 2.48 and Section 2.49.

 LPRINT assumes a 132-character-wide printer.

NOTE: LPRINT and LLIST are not included in all implementations of ANSI BASIC.

2.36 LSET AND RSET

Format: LSET <string variable> = <string expression>
 RSET <string variable> = <string expression>

Purpose: To move data from memory to a random file buffer
(in preparation for a PUT statement).

Remarks: If <string expression> requires fewer bytes than
were FIELDed to <string variable>, LSET
left-justifies the string in the field, and RSET
right-justifies the string. (Spaces are used to
pad the extra positions.) If the string is too
long for the field, characters are dropped from
the right. Numeric values must be converted to
strings before they are LSET or RSET. See the
MKIS\$, MKSS\$, MKDS\$ functions, Section 3.27.

Examples: 150 LSET A\$=MKSS(AMT)
 160 LSET D\$=DESC(\$)

See also Appendix B.

NOTE: LSET or RSET may also be used with a non-fielded
string variable to left-justify or right-justify
a string in a given field. For example, the
program lines

```
110 A$=SPACES$(20)
120 RSET A$=NS
```

right-justify the string NS in a 20-character
field. This can be very handy for formatting
printed output.

2.37 MERGE

Format: MERGE <filename>

Purpose: To merge a specified disk file into the program currently in memory.

Remarks: <filename> is the name used when the file was SAVED. The file must have been SAVED in ASCII format. If not, a "Bad file mode" error occurs.

If any lines in the disk file have the same line numbers as lines in the program in memory, the lines from the file on disk will replace the corresponding lines in memory. (MERGEing may be thought of as "inserting" the program lines on disk into the program in memory.)

ANSI BASIC always returns to command level after executing a MERGE command.

Example: MERGE "NUMBRS"

2.38 MIDS

Format: `MIDS(<string exp1>,n[,m])=<string exp2>`
where `n` and `m` are integer expressions and
`<string exp1>` and `<string exp2>` are string
expressions.

Purpose: To replace a portion of one string with another
string.

Remarks: The characters in `<string exp1>`, beginning at
position `n`, are replaced by the characters in
`<string exp2>`. The optional `m` refers to the
number of characters from `<string exp2>` that
will be used in the replacement. If `m` is
omitted, all of `<string exp2>` is used. However,
regardless of whether `m` is omitted or included,
the replacement of characters never goes beyond
the original length of `<string exp1>`.

Example: `10 A$="KANSAS CITY, MO"`
`20 MIDS(A$,14)="KS"`
`30 PRINT A$`
`RUN`
`KANSAS CITY, KS`

MIDS may also be used as a function that returns
a substring of a given string. See Section
3.26.

2.39 NAME

Format: NAME <old filename> AS <new filename>

Purpose: To change the name of a disk file.

Remarks: <old filename> must exist and <new filename> must not exist; otherwise an error will result. After a NAME command, the file exists on the same disk, in the same area of disk space, with the new name.

Example: Ok
NAME "ACCTS" AS "LEDGER"
Ok

In this example, the file that was formerly named ACCTS will now be named LEDGER.

2.40 NEW

Format: NEW

Purpose: To delete the program currently in memory and clear all variables.

Remarks: NEW is entered at command level to clear memory before entering a new program. ANSI BASIC always returns to command level after a NEW is executed.

2.41 NULL

Format: **NULL <integer expression>**

Purpose: To set the number of nulls to be printed at the end of each line.

Remarks: For 10-character-per-second tape punches, <integer expression> should be ≥ 3 . When tapes are not being punched, <integer expression> should be 0 or 1 for Teletypes and Teletype-compatible CRTs. <integer expression> should be 2 or 3 for 30 cps hard copy printers. The default value is 0.

Example: Ok
NULL 2
Ok
100 INPUT X
200 IF X<50 GOTO 800
.
.
.

Two null characters will be printed after each line.

2.42 ON ERROR GOTO

Format: ON ERROR GOTO <line number>

Purpose: To enable error trapping and specify the first line of the error handling subroutine.

Remarks: Once error trapping has been enabled all errors detected, including direct mode errors (e.g., Syntax errors), will cause a jump to the specified error handling subroutine. If <line number> does not exist, an "Undefined line" error results. To disable error trapping, execute an ON ERROR GOTO 0. Subsequent errors will print an error message and halt execution. An ON ERROR GOTO 0 statement that appears in an error trapping subroutine causes ANSI BASIC to stop and print the error message for the error that caused the trap. It is recommended that all error trapping subroutines execute an ON ERROR GOTO 0 if an error is encountered for which there is no recovery action.

NOTE: If an error occurs during execution of an error handling subroutine, the BASIC error message is printed and execution terminates. Error trapping does not occur within the error handling subroutine.

Example: 10 ON ERROR GOTO 1000

2.43 ON...GOSUB AND ON...GOTO

Format: ON <expression> GOTO <list of line numbers>

 ON <expression> GOSUB <list of line numbers>

Purpose: To branch to one of several specified line numbers, depending on the value returned when an expression is evaluated.

Remarks: The value of <expression> determines which line number in the list will be used for branching. For example, if the value is three, the third line number in the list will be the destination of the branch. (If the value is a non-integer, the fractional portion is rounded.)

In the ON...GOSUB statement, each line number in the list must be the first line number of a subroutine.

If the value of <expression> is negative, zero or greater than the number of items in the list, an "Illegal function call" error occurs.

Example: 100 ON L-1 GOTO 150,300,320,390

2.44 OPEN

Format: OPEN <mode>,[#]<file number>,<filename>,[<reclen>]

Purpose: To allow I/O to a disk file.

Remarks: A disk file must be OPENed before any disk I/O operation can be performed on that file. OPEN allocates a buffer for I/O to the file and determines the mode of access that will be used with the buffer.

<mode> is a string expression whose first character is one of the following:

O specifies sequential output mode

I specifies sequential input mode

R specifies random input/output mode

<file number> is an integer expression whose value is between one and fifteen. The number is then associated with the file for as long as it is OPEN and is used to refer other disk I/O statements to the file.

<filename> is a string expression containing a name that conforms to your operating system's rules for disk filenames.

<reclen> is an integer expression which, if included, sets the record length for random files. The default record length is 124 bytes.

NOTE: A file can be OPENed for sequential input or random access on more than one file number at a time. A file may be OPENed for output, however, on only one file number at a time.

Example: 10 OPEN "I",2,"INVEN"

See also Appendix B.

2.45 OPTION BASE

Format: OPTION BASE n
 where n is 1 or 0

Purpose: To declare the minimum value for array
 subscripts.

Remarks: The default base is 0. If the statement

 OPTION BASE 1

is executed, the lowest value an array subscript
may have is one.

2.46 OUT

Format: OUT I,J
where I and J are integer expressions in the range 0 to 255.

Purpose: To send a byte to a machine output port.

Remarks: The integer expression I is the port number, and the integer expression J is the data to be transmitted.

Example: 100 OUT 32,100

2.47 POKE

Format: POKE I,J
 where I and J are integer expressions

Purpose: To write a byte into a memory location.

Remarks: The integer expression I is the address of the memory location to be POKEd. The integer expression J is the data to be POKEd. J must be in the range 0 to 255. I must be in the range of 0 to 255.

The complementary function to POKE is PEEK. The argument to PEEK is an address from which a byte is to be read. See Section 3.29.

POKE and PEEK are useful for efficient data storage, loading assembly language subroutines, and passing arguments and results to and from assembly language subroutines.

Example: 10 POKE &H5A00,&HFF

2.48 PRINT

Format: PRINT [<list of expressions>]

Purpose: To output data at the terminal.

Remarks: If <list of expressions> is omitted, a blank line is printed. If <list of expressions> is included, the values of the expressions are printed at the terminal. The expressions in the list may be numeric and/or string expressions. (Strings must be enclosed in quotation marks.)

Print Positions

The position of each printed item is determined by the punctuation used to separate the items in the list. ANSI BASIC divides the line into print zones of 14 spaces each. In the list of expressions, a comma causes the next value to be printed at the beginning of the next zone. A semicolon causes the next value to be printed immediately after the last value. Typing one or more spaces between expressions has the same effect as typing a semicolon.

If a comma or a semicolon terminates the list of expressions, the next PRINT statement begins printing on the same line, spacing accordingly.

If the list of expressions terminates without a comma or a semicolon, a carriage return is printed at the end of the line. If the printed line is longer than the terminal width, ANSI BASIC goes to the next physical line and continues printing.

Printed numbers are always followed by a space. Positive numbers are preceded by a space. Negative numbers are preceded by a minus sign. Single precision numbers that can be represented with 6 or fewer digits in the unscaled format no less accurately than they can be represented in the scaled format, are output using the unscaled format. For example, 10^{-6} is output as .000001 and 10^{-7} is output as 1E-7. Double precision numbers that can be represented with 16 or fewer digits in the unscaled format no less accurately than they can be represented in the scaled format, are output using the unscaled format. For example, 10^{-16} is output as .0000000000000001 and 10^{-17} is output as 1D-17.

A question mark may be used in place of the word PRINT in a PRINT statement.

Example 1: 10 X=5
20 PRINT X+5, X-5, X*(-5), X^5
30 END
RUN
10 0 -25 3125
Ok

In this example, the commas in the PRINT statement cause each value to be printed at the beginning of the next print zone.

Example 2: LIST
10 INPUT X
20 PRINT X "SQUARED IS" X^2 "AND";
30 PRINT X "CUBED IS" X^3
40 PRINT
50 GOTO 10
Ok

RUN
? 9
9 SQUARED IS 81 AND 9 CUBED IS 729
? 21
21 SQUARED IS 441 AND 21 CUBED IS 9261
?

In this example, the semicolon at the end of line 20 causes both PRINT statements to be printed on the same line, and line 40 causes a blank line to be printed before the next prompt.

Example 3: 10 FOR X = 1 TO 5
20 J=J+5;
30 K=K+10
40 ?J;K;
50 NEXT X
Ok

RUN
5 10 10 20 15 30 20 40 25 50
Ok

In this example, the semicolons in the PRINT statement cause each value to be printed immediately after the preceding value. (Don't forget, a number is always followed by a space and positive numbers are preceded by a space.) In line 40, a question mark is used instead of the word PRINT.

2.49 PRINT USING

Format: PRINT USING <"format string">;<list of expressions>

Purpose: To print strings or numbers using a specified format.

Remarks and
Examples: <list of expressions> is comprised of the string expressions or numeric expressions that are to be printed, separated by semicolons. <"format

string">, enclosed in quotation marks, is comprised of special formatting characters. These formatting characters (see below) determine the field and the format of the printed strings or numbers.

String Fields

When PRINT USING is used to print strings, one of three formatting characters may be used to format the string field:

"!" Specifies that only the first character in the given string is to be printed.

"\n spaces\" Specifies that 2+n characters from the string are to be printed. If the backslashes are typed with no spaces, two characters will be printed; with one space, three characters will be printed, and so on. If the string is longer than the field, the extra characters are ignored. If the field is longer than the string, the string will be left-justified in the field and padded with spaces on the right.
Example:

```
10 A$="LOOK":B$="OUT"
30 PRINT USING "!" ;A$;B$
40 PRINT USING "\  \" ;A$;B$
50 PRINT USING "\  \" ;A$;B$;"!!"
RUN
LO
LOOKOUT
LOOK OUT  !!
```

"&" Specifies a variable length string field. When the field is specified with "&", the string is output exactly as input. Example:

```
10 A$="LOOK":B$="OUT"
20 PRINT USING "!" ;A$;
30 PRINT USING "&" ;B$
RUN
LOUT
```

Numeric Fields

When PRINT USING is used to print numbers, the following special characters may be used to format the numeric field:

- # A number sign is used to represent each digit position. Digit positions are always filled. If the number to be printed has fewer digits than positions specified, the number will be right-justified (preceded by spaces) in the field.
- . A decimal point may be inserted at any position in the field. If the format string specifies that a digit is to precede the decimal point, the digit will always be printed (as 0 if necessary). Numbers are rounded as necessary.

```
PRINT USING "##.##";.78
0.78
```

```
PRINT USING "###.##";987.654
987.65
```

```
PRINT USING "##.##    ";10.2,5.3,66.789,.234
10.20    5.30    66.79    0.23
```

In the last example, three spaces were inserted at the end of the format string to separate the printed values on the line.

- + A plus sign at the beginning or end of the format string will cause the sign of the number (plus or minus) to be printed before or after the number.

- A minus sign at the end of the format field will cause negative numbers to be printed with a trailing minus sign.

```
PRINT USING "+##.##    "; -68.95, 2.4, 55.6, -.9  
-68.95      +2.40      +55.60      -0.90
```

```
PRINT USING "##.##-   "; -68.95, 22.449, -7.01  
68.95-      22.45      7.01-
```

- ** A double asterisk at the beginning of the format string causes leading spaces in the numeric field to be filled with asterisks. The ** also specifies positions for two more digits.

```
PRINT USING "***#.#   "; 12.39, -0.9, 765.1  
*12.4      *-0.9      765.1
```

- \$\$ A double dollar sign causes a dollar sign to be printed to the immediate left of the formatted number. The \$\$ specifies two more digit positions, one of which is the dollar sign. The exponential format cannot be used with \$\$.
- Negative numbers cannot be used unless the minus sign trails to the right.

```
PRINT USING "$$##.##"; 456.78  
$456.78
```

- ***\$ The ***\$ at the beginning of a format string combines the effects of the above two symbols. Leading spaces will be asterisk-filled and a dollar sign will be printed before the number.
- ***\$ specifies three more digit positions, one of which is the dollar sign.

```
PRINT USING "***$#.##"; 2.34  
***$2.34
```

- ,
- A comma that is to the left of the decimal point in a formatting string causes a comma to be printed to the left of every third digit to the left of the decimal point. A comma that is at the end of the format string is printed as part of the string. A comma specifies another digit position. The comma has no effect if used with the exponential (^^^) format.

```
PRINT USING "###,.##"; 1234.5  
1,234.50
```

```
PRINT USING "###.##,"; 1234.5  
1234.50,
```

~~~~~

Four carats (or up-arrows) may be placed after the digit position characters to specify exponential format. The four carats allow space for E+xx to be printed. Any decimal point position may be specified. The significant digits are left-justified, and the exponent is adjusted. Unless a leading + or trailing + or - is specified, one digit position will be used to the left of the decimal point to print a space or a minus sign.

```
PRINT USING "##.##^^^^";234.56  
2.35E+02
```

```
PRINT USING ".####^^^";888888  
.8889E+06
```

```
PRINT USING "+.##^^^";123  
+.12E+03
```

-

An underscore in the format string causes the next character to be output as a literal character.

```
PRINT USING "_!##.##_!";12.34  
!12.34!
```

The literal character itself may be an underscore by placing "__" in the format string.

%

If the number to be printed is larger than the specified numeric field, a percent sign is printed in front of the number. If rounding causes the number to exceed the field, a percent sign will be printed in front of the rounded number.

```
PRINT USING "##.##";111.22  
%111.22
```

```
PRINT USING ".##";.999  
%1.00
```

If the number of digits specified exceeds 24, an "Illegal function call" error will result.

2.50 PRINT# AND PRINT# USING

Format: PRINT#<filenumber>,[USING<"format string">];<list of exps>

Purpose: To write data to a sequential disk file.

Remarks: <file number> is the number used when the file was OPENed for output. <"format string"> is comprised of formatting characters as described in Section 2.49, PRINT USING. The expressions in <list of expressions> are the numeric and/or string expressions that will be written to the file.

PRINT# does not compress data on the disk. An image of the data is written to the disk, just as it would be displayed on the terminal with a PRINT statement. For this reason, care should be taken to delimit the data on the disk, so that it will be input correctly from the disk.

In the list of expressions, numeric expressions should be delimited by semicolons. For example, PRINT#1,A;B;C;X;Y;Z

(If commas are used as delimiters, the extra blanks that are inserted between print fields will also be written to disk.)

String expressions must be separated by semicolons in the list. To format the string expressions correctly on the disk, use explicit delimiters in the list of expressions.

For example, let A\$="CAMERA" and B\$="93604-1". The statement

PRINT#1,A\$;B\$

would write CAMERA93604-1 to the disk. Because there are no delimiters, this could not be input as two separate strings. To correct the problem, insert explicit delimiters into the PRINT# statement as follows:

PRINT#1,A\$;",";B\$

The image written to disk is

CAMERA,93604-1

which can be read back into two string variables.

If the strings themselves contain commas, semicolons, significant leading blanks, carriage returns, or line feeds, write them to disk surrounded by explicit quotation marks, CHR\$(34).

For example, let A\$="CAMERA, AUTOMATIC" and B\$=" 93604-1". The statement
PRINT#1,A\$;B\$

would write the following image to disk:

CAMERA, AUTOMATIC 93604-1

and the statement

INPUT#1,A\$,B\$

would input "CAMERA" to A\$ and "AUTOMATIC 93604-1" to B\$. To separate these strings properly on the disk, write double quotes to the disk image using CHR\$(34). The statement

PRINT#1,CHR\$(34);A\$;CHR\$(34);CHR\$(34);B\$;CHR\$(34)

writes the following image to disk:

"CAMERA, AUTOMATIC"" 93604-1"

and the statement

INPUT#1,A\$,B\$

would input "CAMERA, AUTOMATIC" to A\$ and " 93604-1" to B\$.

The PRINT# statement may also be used with the USING option to control the format of the disk file. For example:

PRINT#1,USING"####.##,";J;K;L

For more examples using PRINT#, see Appendix B.

See also WRITE#, Section 2.68.

2.51 PUT

Format: PUT [#]<file number>[,<record number>]

Purpose: To write a record from a random buffer to a random disk file.

Remarks: <file number> is the number under which the file was OPENed. If <record number> is omitted, the record will have the next available record number (after the last PUT). The largest possible record number is 32767.

Example: See Appendix B.

2.52 RANDOMIZE

Format: RANDOMIZE [<expression>]

Purpose: To reseed the random number generator.

Remarks: If <expression> is omitted, ANSI BASIC suspends program execution and asks for a value by printing

Random Number Seed (0-65529)?

before executing RANDOMIZE.

If the random number generator is not reseeded, the RND function returns the same sequence of random numbers each time the program is RUN. To change the sequence of random numbers every time the program is RUN, place a RANDOMIZE statement at the beginning of the program and change the argument with each RUN.

Example:

```
10 RANDOMIZE
20 FOR I=1 TO 5
30 PRINT RND;
40 NEXT I
RUN
Random Number Seed (0-65529)? 3 (user types 3)
  .88598  .484668  .586328  .119426  .709225
Ok
RUN
Random Number Seed (0-65529)? 4 (user types 4
for new sequence)
  .803506  .162462  .929364  .292443  .322921
Ok
RUN
Random Number Seed (0-65529)? 3 (same sequence
as first RUN)
  .88598  .484668  .586328  .119426  .709225
Ok
```

2.53 READ

Format: READ <list of variables>

Purpose: To read values from a DATA statement and assign them to variables. (See DATA, Section 2.9.)

Remarks: A READ statement must always be used in conjunction with a DATA statement. READ statements assign variables to DATA statement values on a one-to-one basis. READ statement variables may be numeric or string, and the values read must agree with the variable types specified. If they do not agree, a "Syntax error" will result.

A single READ statement may access one or more DATA statements (they will be accessed in order), or several READ statements may access the same DATA statement. If the number of variables in <list of variables> exceeds the number of elements in the DATA statement(s), an OUT OF DATA message is printed. If the number of variables specified is fewer than the number of elements in the DATA statement(s), subsequent READ statements will begin reading data at the first unread element. If there are no subsequent READ statements, the extra data is ignored.

To reread DATA statements from the start, use the RESTORE statement (see RESTORE, Section 2.57)

Example 1:

```
.  
. .  
.  
80 FOR I=1 TO 10  
90 READ A(I)  
100 NEXT I  
110 DATA 3.08,5.19,3.12,3.98,4.24  
120 DATA 5.08,5.55,4.00,3.16,3.37  
. .  
. .
```

This program segment READs the values from the DATA statements into the array A. After execution, the value of A(1) will be 3.08, and so on.

Example 2: LIST

```
10 PRINT "CITY", "STATE", " ZIP"
20 READ CS,SS,Z
30 DATA "DENVER,", COLORADO, 80211
40 PRINT CS,SS,Z
Ok
RUN
CITY           STATE          ZIP
DENVER,        COLORADO      80211
Ok
```

This program READs string and numeric data from the DATA statement in line 30.

2.54 REM

Format: REM <remark>

Purpose: To allow explanatory remarks to be inserted in a program.

Remarks: REM statements are not executed but are output exactly as entered when the program is listed.

REM statements may be branched into (from a GOTO or GOSUB statement), and execution will continue with the first executable statement after the REM statement.

In MOSTEK ANSI BASIC, remarks may be added to the end of a line by preceding the remark with a single quotation mark instead of :REM.

Example:

```
120 REM CALCULATE AVERAGE VELOCITY  
130 FOR I=1 TO 20  
140 SUM=SUM + V(I)
```

or, with MOSTEK ANSI BASIC version:

```
120 FOR I=1 TO 20      'CALCULATE AVERAGE VELOCITY  
130 SUM=SUM+V(I)  
140 NEXT I  
.  
.  
.
```

2.55 RENUM

Format: RENUM { [*<new number>*] [, [*<old number>*] [, *<increment>*]] }

Purpose: To renumber program lines.

Remarks: <new number> is the first line number to be used in the new sequence. The default is 10. <old number> is the line in the current program where renumbering is to begin. The default is the first line of the program. <increment> is the increment to be used in the new sequence. The default is 10.

RENUM also changes all line number references following GOTO, GOSUB, THEN, ON...GOTO, ON...GOSUB and ERL statements to reflect the new line numbers. If a nonexistent line number appears after one of these statements, the error message "Undefined line xxxxx in yyyy" is printed. The incorrect line number reference (xxxxx) is not changed by RENUM, but line number yyyy may be changed.

NOTE: RENUM cannot be used to change the order of program lines (for example, RENUM 15,30 when the program has three lines numbered 10, 20 and 30) or to create line numbers greater than 65529. An "Illegal function call" error will result.

RENUM 300,,50 Renumbers the entire program. The first new line number will be 300. Lines will increment by 50.

RENUM 1000,900,20 Renumbers the lines from 900 up so they start with line number 1000 and increment by 20.

2.56 RESET

Format: RESET

Purpose: To close all disk files and write the directory information from a diskette to memory before any disk I/O operation begins.

Remarks: Always execute the RESET command when changing diskettes on a disk unit. The RESET command is executable both in direct and indirect mode. It can be used to access a database residing in more than one diskette.

NOTE

In MOSTEK ANSI BASIC, RESET initializes all disk units. When ANSI BASIC is entered all disks are automatically initialized.

2.57 RESTORE

Format: RESTORE [<line number>]

Purpose: To allow DATA statements to be reread from a specified point.

Remarks: After a RESTORE statement is executed, the next READ statement accesses the first item in the first DATA statement in the program. If <line number> is specified, the next READ statement accesses the first item in the specified DATA statement.

Example: 10 READ A,B,C

```
20 RESTORE
30 READ D,E,F
40 DATA 57, 68, 79
```

```
    .
    .
    .
```

2.58 RESUME

Formats: RESUME
RESUME 0
RESUME NEXT
RESUME <line number>

Purpose: To continue program execution after an error recovery procedure has been performed.

Remarks: Any one of the four formats shown above may be used, depending upon where execution is to resume:

RESUME Execution resumes at the
 or
RESUME 0 statement which caused the
 error.

RESUME NEXT Execution resumes at the statement immediately following the one which caused the error.

RESUME <line number> Execution resumes at
<line number>.

A RESUME statement that is not in an error trap routine causes a "RESUME without error" message to be printed.

Example: 10 ON ERROR GOTO 900

1

1

1

900 IF (ERR=230)AND(ERL=90) THEN PRINT "TRY AGAIN":RESUME 80

1

1

1

2.59 RUN

Format 1: RUN [<line number>]

Purpose: To execute the program currently in memory.

Remarks: If <line number> is specified, execution begins on that line. Otherwise, execution begins at the lowest line number. BASIC always returns to command level after a RUN is executed.

Example: RUN

Format 2: RUN <filename>[,R]

Purpose: To load a file from disk into memory and run it.

Remarks: <filename> is the name used when the file was SAVED.

RUN closes all open files and deletes the current contents of memory before loading the designated program. However, with the "R" option, all data files remain OPEN.

Example: RUN "NEWFIL",R

See also Appendix B.

2.60 SAVE

Format: `SAVE <filename>[,A ° ,P]`

Purpose: To save a program file on disk.

Remarks: `<filename>` is a quoted string that conforms to your operating system's requirements for filenames. If `<filename>` already exists, then the file will be written over.

Use the A option to save the file in ASCII format. Otherwise, BASIC saves the file in a compressed binary format. ASCII format takes more space on the disk, but some disk access requires that files be in ASCII format. For instance, the MERGE command requires an ASCII format file, and some operating system commands such as LIST may require an ASCII format file.

Use the P option to protect the file by saving it in an encoded binary format. When a protected file is later RUN (or LOADED), any attempt to list or edit it will fail.

Examples: `SAVE"COM2",A`
 `SAVE"PROG",P`

See also Appendix B.

2.61 STOP

Format: STOP

Purpose: To terminate program execution and return to command level.

Remarks: STOP statements may be used anywhere in a program to terminate execution. When a STOP is encountered, the following message is printed:

Break in line nnnn

Unlike the END statement, the STOP statement does not close files.

BASIC always returns to command level after a STOP is executed. Execution is resumed by issuing a CONT command (see Section 2.7).

Example:

```
10 INPUT A,B,C
20 K=A^2*5.3:L=B^3/.26
30 STOP
40 M=C*K+100:PRINT M
RUN
? 1,2,3
BREAK IN 30
Ok
PRINT L
 30.7692
Ok
CONT
 115.9
Ok
```

2.62 SWAP

Format: SWAP <variable>,<variable>

Purpose: To exchange the values of two variables.

Remarks: Any type variable may be SWAPPED (integer, single precision, double precision, string), but the two variables must be of the same type or a "Type mismatch" error results.

Example: LIST

```
10 A$=" ONE " : B$=" ALL " : C$="FOR"
20 PRINT A$ C$ B$
30 SWAP A$, B$
40 PRINT A$ C$ B$
RUN
Ok
    ONE FOR ALL
    ALL FOR ONE
Ok
```

2.62 SYSTEM

Format: SYSTEM

Purpose: To return to the operating system from BASIC.

Remarks: All opened files are closed and the operating system is rebooted from DKO: .

2.63 TRON/TROFF

Format: TRON
TROFF

Purpose: To trace the execution of program statements.

Remarks: As an aid in debugging, the TRON statement (executed in either the direct or indirect mode) enables a trace flag that prints each line number of the program as it is executed. The numbers appear enclosed in square brackets. The trace flag is disabled with the TROFF statement (or when a NEW command is executed).

Example: TRON

```
Ok
LIST
10 K=10
20 FOR J=1 TO 2
30 L=K + 10
40 PRINT J;K;L
50 K=K+10
60 NEXT
70 END
Ok
RUN
[10][20][30][40] 1 10 20
[50][60][30][40] 2 20 30
[50][60][70]
Ok
TROFF
Ok
```

2.64 WAIT

Format: WAIT <port number>, I[,J]
 where I and J are integer expressions

Purpose: To suspend program execution while monitoring
 the status of a machine input port.

Remarks: The WAIT statement causes execution to be suspended until a specified machine input port develops a specified bit pattern. The data read at the port is exclusive OR'ed with the integer expression J, and then AND'ed with I. If the result is zero, ANSI BASIC loops back and reads the data at the port again. If the result is nonzero, execution continues with the next statement. If J is omitted, it is assumed to be zero

CAUTION: It is possible to enter an infinite loop with the WAIT statement, in which case it will be necessary to manually restart the machine.

Example: 100 WAIT 32,2

2.65 WHILE...WEND

Format: WHILE <expression>
 .
 .
 [<loop statements>]
 .
 .
 WEND

Purpose: To execute a series of statements in a loop as long as a given condition is true.

Remarks: If <expression> is not zero (i.e., true), <loop statements> are executed until the WEND statement is encountered. BASIC then returns to the WHILE statement and checks <expression>. If it is still true, the process is repeated. If it is not true, execution resumes with the statement following the WEND statement.

WHILE/WEND loops may be nested to any level. Each WEND will match the most recent WHILE. An unmatched WHILE statement causes a "WHILE without WEND" error, and an unmatched WEND statement causes a "WEND without WHILE" error.

Example:

```
90 'BUBBLE SORT ARRAY A$  
100 FLIPS=1 'FORCE ONE PASS THRU LOOP  
110 WHILE FLIPS  
115         FLIPS=0  
120         FOR I=1 TO J-1  
130             IF A$(I)>A$(I+1) THEN  
                       SWAP A$(I),A$(I+1):FLIPS=1  
140         NEXT I  
150 WEND
```

2.66 WIDTH

Format: WIDTH [LPRINT] <integer expression>

Purpose: To set the printed line width in number of characters for the terminal or line printer.

Remarks: If the LPRINT option is omitted, the line width is set at the terminal. If LPRINT is included, the line width is set at the line printer.

<integer expression> must have a value in the range 15 to 255. The default width is 72 characters.

If <integer expression> is 255, the line width is "infinite," that is, BASIC never inserts a carriage return. However, the position of the cursor or the print head, as given by the POS or LPOS function, returns to zero after position 255.

Example: 10 PRINT "ABCDEFGHIJKLMNOPQRSTUVWXYZ"

```
RUN
ABCDEFGHIJKLMNOPQRSTUVWXYZ
Ok
WIDTH 18
Ok
RUN
ABCDEFGHIJKLMNOPQR
STUVWXYZ
Ok
```

2.67 WRITE

Format: `WRITE[<list of expressions>]`

Purpose: To output data at the terminal.

Remarks: If `<list of expressions>` is omitted, a blank line is output. If `<list of expressions>` is included, the values of the expressions are output at the terminal. The expressions in the list may be numeric and/or string expressions, and they must be separated by commas.

When the printed items are output, each item will be separated from the last by a comma. Printed strings will be delimited by quotation marks. After the last item in the list is printed, BASIC inserts a carriage return/line feed.

WRITE outputs numeric values using the same format as the PRINT statement, Section 2.48.

Example: `10 A=80:B=90:C$="THAT'S ALL"`
 `20 WRITE A,B,CS`
 `RUN`
 `80, 90,"THAT'S ALL"`
 `Ok`

2.68 WRITE#

Format: `WRITE#<file number>,<list of expressions>`

Purpose: To write data to a sequential file.

Remarks: `<file number>` is the number under which the file was OPENed in "0" mode. The expressions in the list are string or numeric expressions, and they must be separated by commas.

The difference between `WRITE#` and `PRINT#` is that `WRITE#` inserts commas between the items as they are written to disk and delimits strings with quotation marks. Therefore, it is not necessary for the user to put explicit delimiters in the list. A carriage return/line feed sequence is inserted after the last item in the list is written to disk.

Example: Let `A$="CAMERA"` and `B$="93604-1"`. The statement:

`WRITE#1,A$,B$`

writes the following image to disk:

`"CAMERA","93604-1"`

A subsequent `INPUT#` statement, such as:

`INPUT#1,A$,B$`

would input "CAMERA" to `A$` and "93604-1" to `B$`.

CHAPTER 3

ANSI BASIC FUNCTIONS

The intrinsic functions provided by ANSI BASIC are presented in this chapter. The functions may be called from any program without further definition.

Arguments to functions are always enclosed in parentheses. In the formats given for the functions in this chapter, the arguments have been abbreviated as follows:

X and Y	Represent any numeric expressions
I and J	Represent integer expressions
X\$ and Y\$	Represent string expressions

If a floating point value is supplied where an integer is required, ANSI BASIC will round the fractional portion and use the resulting integer.

3.1 ABS

Format: ABS(X)

Action: Returns the absolute value of the expression X.

Example: PRINT ABS(7*(-5))
35
Ok

3.2 ASC

Format: ASC(X\$)

Action: Returns a numerical value that is the ASCII code of the first character of the string X\$. (See Appendix L for ASCII codes.) If X\$ is null, an "Illegal function call" error is returned.

Example: 10 X\$ = "TEST"
20 PRINT ASC(X\$)
RUN
84
Ok
See the CHR\$ function for ASCII-to-string conversion.

3.3 ATN

Format: ATN(X)

Action: Returns the arctangent of X in radians. Result is in the range -pi/2 to pi/2. The expression X may be any numeric type, but the evaluation of ATN is always performed in single precision.

Example: 10 INPUT X
20 PRINT ATN(X)
RUN
? 3
1.24905
Ok

3.4 CDBL

Format: CDBL(X)

Action: Converts X to a double precision number.

Example: 10 A = 454.67
20 PRINT A;CDBL(A)
RUN
454.67 454.6700134277344
Ok

3.5 CHR\$

Format: CHR\$(I)

Action: Returns a string whose one element has ASCII code I. (ASCII codes are listed in Appendix G.) CHR\$ is commonly used to send a special character to the terminal. For instance, the BEL character could be sent (CHR\$(7)) as a preface to an error message, or a form feed could be sent (CHR\$(12)) to clear a CRT screen and return the cursor to the home position.

Example: PRINT CHR\$(66)
B
Ok

See the ASC function for ASCII-to-numeric conversion.

3.6 CINT

Format: CINT(X)

Action: Converts X to an integer by rounding the fractional portion. If X is not in the range -32768 to 32767, an "Overflow" error occurs.

Example: PRINT CINT(45.67)
46
Ok

See the CDBL and CSNG functions for converting numbers to the double precision and single precision data type. See also the FIX and INT functions, both of which return integers.

3.7 COS

Format: COS(X)

Action: Returns the cosine of X in radians. The calculation of COS(X) is performed in single precision.

Example:

```
10 X = 2*COS(.4)
20 PRINT X
RUN
1.84212
Ok
```

3.8 CSNG

Format: CSNG(X)

Action: Converts X to a single precision number.

Example:

```
10 A# = 975.3421#
20 PRINT A#; CSNG(A#)
RUN
975.3421 975.342
Ok
```

See the CINT and CDBL functions for converting numbers to the integer and double precision data types.

3.9 CVI, CVS, CVD

Format: CVI(<2-byte string>)
 CVS(<4-byte string>)
 CVD(<8-byte string>)

Action: Convert string values to numeric values. Numeric values that are read in from a random disk file must be converted from strings back into numbers. CVI converts a 2-byte string to an integer. CVS converts a 4-byte string to a single precision number. CVD converts an 8-byte string to a double precision number.

Example: .
 .
 .
 .
70 FIELD #1,4 AS N\$, 12 AS B\$, ...
80 GET #1
90 Y=CVS(N\$)
 .
 .
 .

See also MKIS, MKSS, MKDS, Section 3.25 and Appendix B.

3.10 DSKF

Format: DSKF(<disk unit number>)

Action: Returns the number of free sectors available in the specified disk unit.

Example: 100 IF DSKF(0)=0 THEN 900
110 PUT #1

3.11 EOF

Format: EOF(<file number>)

Action: Returns -1 (true) if the end of a sequential file has been reached. Use EOF to test for end-of-file while INPUTting, to avoid "Input past end" errors.

Example: 10 OPEN "I",1,"DATA"
 20 C=0
 30 IF EOF(1) THEN 100
 40 INPUT #1,M(C)
 50 C=C+1:GOTO 30

3.12 EXP

Format: EXP(X)

Action: Returns e to the power of X. X must be <=87.3365. If EXP overflows, the "Overflow" error message is displayed, machine infinity with the appropriate sign is supplied as the result, and execution continues.

Example: 10 X = 5
 20 PRINT EXP (X-1)
 RUN
 54.5982
 Ok

3.13 FIX

Format: FIX(X)

Action: Returns the truncated integer part of X. FIX(X) is equivalent to SGN(X)*INT(ABS(X)). The major difference between FIX and INT is that FIX does not return the next lower number for negative X.

Examples: PRINT FIX(58.75)
 58
 Ok
 PRINT FIX(-58.75)
 -58
 Ok

3.14 FRE

Format: FRE(0)

FRE(X\$)

Action: Arguments to FRE are dummy arguments. If the argument is 0 (numeric), FRE returns the number of bytes in memory not being used by ANSI BASIC. If the argument is a string, FRE returns the number of free bytes in string space.

Example: PRINT FRE(0)
14542
Ok

3.15 HEX\$

Format: HEX\$(X)

Action: Returns a string which represents the hexadecimal value of the decimal argument. X is rounded to an integer before HEX\$(X) is evaluated.

Example: 10 INPUT X
20 AS = HEX\$(X)
30 PRINT X "DECIMAL IS " AS " HEXADECIMAL"
RUN
? 32
32 DECIMAL IS 20 HEXADECIMAL
Ok

See the OCT\$ function for octal conversion.

3.16 INP

Format: INP(I)

Action: Returns the byte read from port I. I must be in the range 0 to 255. INP is the complementary function to the OUT statement, Section 2.46.

Example: 100 A=INP(255)

3.17 INPUT\$

Format: INPUT\$(X[,#]Y)

Action: Returns a string of X characters, read from the terminal or from file number Y. If the terminal is used for input, no characters will be echoed and all control characters are passed through except Control-C, which is used to interrupt the execution of the INPUT\$ function.

Example 1: 5 'LIST THE CONTENTS OF A SEQUENTIAL FILE IN
HEXADECIMAL
10 OPEN "I", 1, "DATA"
20 IF EOF(1) THEN 50
30 PRINT HEX\$(ASC(INPUT\$(1,#1)));
40 GOTO 20
50 PRINT
60 END

Example 2: .
. .
100 PRINT "TYPE P TO PROCEED OR S TO STOP"
110 X\$=INPUT\$(1)
120 IF X\$="P" THEN 500
130 IF X\$="S" THEN 700 ELSE 100
. . .

3.18 INSTR

Format: INSTR([I,]X\$,Y\$)

Action: Searches for the first occurrence of string Y\$ in X\$ and returns the position at which the match is found. Optional offset I sets the position for starting the search. I must be in the range 0 to 255. If I>LEN(X\$) or if X\$ is null or if Y\$ cannot be found, INSTR returns 0. If Y\$ is null, INSTR returns I or 1. X\$ and Y\$ may be string variables, string expressions or string literals.

Example: 10 XS = "ABCDEB"
20 Y\$ = "B"
30 PRINT INSTR(X\$,Y\$);INSTR(4,X\$,Y\$)
RUN
2 6
Ok

3.19 INT

Format: INT(X)

Action: Returns the largest integer <=X.

Examples: PRINT INT(99.89)
99
Ok
PRINT INT(-12.11)
-13
Ok

See the FIX and CINT functions which also return integer values.

3.20 LEFT\$

Format: LEFT\$(X\$,I)

Action: Returns a string comprised of the leftmost I characters of X\$. I must be in the range 0 to 255. If I is greater than LEN(X\$), the entire string (X\$) will be returned. If I=0, the null string (length zero) is returned.

Example: 10 A\$ = "ANSI BASIC"
20 B\$ = LEFT\$(A\$,5)
30 PRINT B\$
BASIC
Ok

Also see the MIDS and RIGHTS functions.

3.21 LEN

Format: LEN(X\$)

Action: Returns the number of characters in X\$. Non-printing characters and blanks are counted.

Example: 10 XS = "PORTLAND, OREGON"
20 PRINT LEN(X\$)
16
Ok

3.22 LOC

Format: LOC(<file number>)

Action: With random disk files, LOC returns the current record number to be used if a GET or PUT (without a record number) is executed. With sequential files, LOC returns the number of sectors (124 byte blocks) read from or written to the file since it was OPENed.

Example: 200 IF LOC(1)>50 THEN STOP

3.23 LOF

Format: LOF(<file number>)

Action: Returns the number of records in the file.

Example: IF LOF(1)=0 THEN 900

3.24 LOG

Format: LOG(X)

Action: Returns the natural logarithm of X. X must be greater than zero.

Example: PRINT LOG(35/7)
1.86075
Ok

3.25 LPOS

Format: LPOS(X)

Action: Returns the current position of the line printer print head within the line printer buffer. Does not necessarily give the physical position of the print head. X is a dummy argument.

Example: 100 IF LPOS(X)>60 THEN LPRINT CHR\$(13)

3.26 MIDS

Format: MIDS(X\$, I[, J])

Action: Returns a string of length J characters from X\$ beginning with the Ith character. I and J must be in the range 0 to 255. If J is omitted or if there are fewer than J characters to the right of the Ith character, all rightmost characters beginning with the Ith character are returned. If I>LEN(X\$), MIDS returns a null string.

Example:

```
LIST
10 A$="GOOD "
20 B$="MORNING EVENING AFTERNOON"
30 PRINT A$;MIDS(B$,9,7)
Ok
RUN
GOOD EVENING
Ok
```

Also see the LEFT\$ and RIGHTS functions.

3.27 MKIS\$, MKSS\$, MKDS\$

Format: MKIS\$(<integer expression>
 MKSS\$(<single precision expression>
 MKDS\$(<double precision expression>)

Action: Convert numeric values to string values. Any numeric value that is placed in a random file buffer with an LSET or RSET statement must be converted to a string. MKIS\$ converts an integer to a 2-byte string. MKSS\$ converts a single precision number to a 4-byte string. MKDS\$ converts a double precision number to an 8-byte string.

Example: 90 AMT=(K+T)
100 FIELD #1, 8 AS D\$, 20 AS N\$
110 LSET D\$ = MKSS(AMT)
120 LSET N\$ = A\$
130 PUT #1
•
•

See also CVI, CVS, CVD, Section 3.9 and Appendix B.

3.28 OCTS\$

Format: OCT\$(X)

Action: Returns a string which represents the octal value of the decimal argument. X is rounded to an integer before OCT\$(X) is evaluated.

Example: PRINT OCT\$(24)
 30
 Ok

See the HEX\$ function for hexadecimal conversion.

3.29 PEEK

Format: PEEK(I)

Action: Returns the byte (decimal integer in the range 0 to 255) read from memory location I. I must be in the range 0 to 65536. PEEK is the complementary function to the POKE statement, Section 2.47.

Example: A=PEEK(&H5A00)

3.30 POS

Format: POS(I)

Action: Returns the current cursor position. The leftmost position is 0. X is a dummy argument.

Example: IF POS(X)>60 THEN PRINT CHR\$(13)
Also see the LPOS function.

3.31 RIGHT\$

Format: RIGHT\$(X\$, I)

Action: Returns the rightmost I characters of string X\$.
If I=LEN(X\$), returns X\$. If I=0, the null
string (length zero) is returned.

Example: 10 A\$="DISK ANSI BASIC"
20 PRINT RIGHT\$(A\$, 8)
RUN
ANSI BASIC
Ok

Also see the MIDS and LEFT\$ functions.

3.32 RND

Format: RND[(X)]

Action: Returns a random number between 0 and 1. The
same sequence of random numbers is generated
each time the program is RUN unless the random
number generator is reseeded (see RANDOMIZE,
Section 2.52). However, X<0 always restarts the
same sequence for any given X.

X>0 or X omitted generates the next random
number in the sequence. X=0 repeats the last
number generated.

Example: 10 FOR I=1 TO 5
20 PRINT INT(RND*100);
30 NEXT
RUN
24 30 31 51 5
Ok

3.33 SGN

Format: SGN(X)

Action: If X>0, SGN(X) returns 1.
 If X=0, SGN(X) returns 0.
 If X<0, SGN(X) returns -1.

Example: ON SGN(X)+2 GOTO 100,200,300 branches to 100 if
 X is negative, 200 if X is 0 and 300 if X is
 positive.

3.34 SIN

Format: SIN(X)

Action: Returns the sine of X in radians. SIN(X) is
 calculated in single precision.
 COS(X)=SIN(X+3.14159/2).

Example: PRINT SIN(1.5)
 .997495
 Ok

3.35 SPACES\$

Format: SPACES\$(X)

Action: Returns a string of spaces of length X. The expression X is rounded to an integer and must be in the range 0 to 255.

Example: 10 FOR I = 1 TO 5
20 XS = SPACES\$(I)
30 PRINT XS;I
40 NEXT I
RUN
1
2
3
4
5
Ok

Also see the SPC function.

3.36 SPC

Format: SPC(I)

Action: Prints I blanks on the terminal. SPC may only be used with PRINT and LPRINT statements. I must be in the range 0 to 255.

Example: PRINT "OVER" SPC(15) "THERE"
OVER THERE
Ok

Also see the SPACES\$ function.

3.37 SQR

Format: SQR(X)

Action: Returns the square root of X. X must be ≥ 0 .

Example: 10 FOR X = 10 TO 25 STEP 5
20 PRINT X, SQR(X)
30 NEXT
RUN
10 3.16228
15 3.87298
20 4.47214
25 5
Ok

3.38 STR\$

Format: STR\$(X)

Action: Returns a string representation of the value of X.

Example: 5 REM ARITHMETIC FOR KIDS
10 INPUT "TYPE A NUMBER";N
20 ON LEN(STRS(N)) GOSUB 30,100,200,300,400,500
•
•
•

Also see the VAL function.

3.39 STRINGS

Formats: **STRINGS(I,J)**

STRINGS(I,X\$)

Action: Returns a string of length I whose characters all have ASCII code J or the first character of XS.

```
Example:    10 X$ = STRINGS(10,45)
20 PRINT X$ "MONTHLY REPORT" X$
RUN
-----MONTHLY REPORT-----
Ok
```

3.40 TAB

Format: TAB(I)

Action: Spaces to position I on the terminal. If the current print position is already beyond space I, TAB has no effect. Space 0 is the leftmost position, and the rightmost position is the width minus one. I must be in the range 0 to 255. TAB may only be used in PRINT and LPRINT statements.

```
Example:    10 PRINT "NAME" TAB(25) "AMOUNT" : PRINT
              20 READ A$,B$
              30 PRINT A$ TAB(25) B$
              40 DATA "G. T. JONES","$25.00"
              RUN
              NAME                 AMOUNT
              G. T. JONES          $25.00
              Ok
```

3.41 TAN

Format: TAN(X)

Action: Returns the tangent of X in radians. TAN(X) is calculated in single precision. If TAN overflows, the "Overflow" error message is displayed, machine infinity with the appropriate sign is supplied as the result, and execution continues.

Example: 10 Y = Q*TAN(X)/2

3.42 USR

Format : USR[<digit>](X)

Action: Calls the user's assembly language subroutine with the argument X. <digit> is in the range of 0-9 and corresponds to the digit supplied with the DEF USR statement for that routine. If <digit> is omitted, USR0 is assumed. See Appendix C.

Example: 40 B = T*SIN(Y)
50 C = USR(B/2)
60 D = USR(B/3)

•
•
•

3.43 VAL

Format: VAL(X\$)

Action: Returns the numerical value of string X\$. If the first character of X\$ is not +, -, &, or a digit, VAL(X\$)=0.

Example:

```
10 READ NAMES,CITY$,STATES,ZIPS
20 IF VAL(ZIPS)<90000 OR VAL(ZIPS)>96699 THEN
   PRINT NAMES TAB(25) "OUT OF STATE"
30 IF VAL(ZIPS)>=90801 AND VAL(ZIPS)<=90815 THEN
   PRINT NAMES TAB(25) "LONG BEACH"
   .
   .
   .
```

See the STR\$ function for numeric to string conversion.

3.44 VARPTR

Format 1: VARPTR(<variable name>)

Format 2: VARPTR(#<file number>)

Version: Disk

Action: Format 1: Returns the address of the first byte of data identified with <variable name>. A value must be assigned to <variable name> prior to execution of VARPTR. Otherwise an "Illegal function call" error results. Any type variable name may be used (numeric, string, array), and the address returned will be an integer in the range 32767 to -32768. If a negative address is returned, add it to 65536 to obtain the actual address.

VARPTR is usually used to obtain the address of a variable or array so it may be passed to an assembly language subroutine. A function call of the form VARPTR(A(0)) is usually specified when passing an array, so that the lowest-addressed element of the array is returned.

NOTE: All simple variables should be assigned before calling VARPTR for an array, because the addresses of the arrays change whenever a new simple variable is assigned.

Format 2: Returns the starting address of the disk I/O buffer assigned to <file number>.

Example: 100 X=USR(VARPTR(Y))

APPENDIX A

New Features in ANSI BASIC, Release 5.3

The execution of BASIC programs written under MOSTEK BASIC release 5.1 or earlier, and Microsoft BASIC, release 4.51 or earlier may be affected by some of the new features in ANSI BASIC. Before attempting to run such programs, check for the following:

1. New reserved words: CALL, CHAIN, COMMON, WHILE, WEND, WRITE, OPTION BASE, RANDOMIZE.
2. Conversion from floating point to integer values results in rounding, as opposed to truncation. This affects not only assignment statements (e.g., I%=2.5 results in I%=3), but also affects function and statement evaluations (e.g., TAB(4.5) goes to the 5th position, A(1.5) yeilds A(2), and X=11.5 MOD 4 yields 0 for X).
3. The body of a FOR...NEXT loop is skipped if the initial value of the loop times the sign of the step exceeds the final value times the sign of the step. See Section 2.21.
4. Division by zero and overflow no longer produce fatal errors. See Section 1.8.1.2.
5. The RND function has been changed so that RND with no argument is the same as RND with a positive argument. The RND function generates the same sequence of random numbers with each RUN, unless RANDOMIZE is used. See Sections 2.52 and 3.32.
6. The rules for PRINTing single precision and double precision numbers have been changed. See Section 2.48.
7. If the argument to ON...GOTO is out of range, an error message results and execution halts.
8. String space is allocated dynamically, and the first argument in a two-argument CLEAR statement will be ignored. See Section 2.4.

9. Responding to INPUT with too many or too few items, or with the wrong type of value (numeric instead of string, etc.), or with a carriage return causes the message "?Redo from start" to be printed. No assignment of input values is made until an acceptable response is given, except when only one variable is given in the INPUT list. See Section 2.26.
10. There are two new field formatting characters for use with PRINT USING. An ampersand is used for variable length string fields, and an underscore signifies a literal character in a format string.
11. If the expression supplied with the WIDTH statement is 255, BASIC uses an "infinite" line width, that is, it does not insert carriage returns. WIDTH LPRINT may be used to set the line width at the line printer. See Section 2.66.
12. The at-sign and underscore are no longer used as editing characters.
13. Variable names are significant up to 40 characters and can contain embedded reserved words. However, reserved words must now be delimited by spaces. To maintain compatibility with earlier versions of BASIC, spaces will be automatically inserted between adjoining reserved words and variable names. WARNING: This insertion of spaces may cause the end of a line to be truncated if the line length is close to 255 characters.
14. BASIC programs may be saved in a protected binary format. See SAVE, Section 2.60.

APPENDIX B

ANSI BASIC Disk I/O

Disk I/O procedures for the beginning ANSI BASIC user are examined in this appendix. If you are new to ANSI BASIC or if you're getting disk related errors, read through these procedures and program examples to make sure you're using all the disk statements correctly.

Wherever a filename is required in a disk command or statement, use a name that conforms to your operating system's requirements for filenames.

B.1 PROGRAM FILE COMMANDS

Here is a review of the commands and statements used in program file manipulation.

SAVE "filename"[,A] Writes to disk the program that is currently residing in memory. Optional A writes the program as a series of ASCII characters. (Otherwise, BASIC uses a compressed binary format.)

LOAD "filename"[,R] Loads the program from disk into memory. Optional R runs the program immediately. LOAD always deletes the current contents of memory and closes all files before LOADING. If R is included, however, open data files are kept open. Thus programs can be chained or loaded in sections and access the same data files.

RUN "filename"[,R]	RUN "filename" loads the program from disk into memory and runs it. RUN deletes the current contents of memory and closes all files before loading the program. If the R option is included, however, all open data files are kept open.
MERGE "filename"	Loads the program from disk into memory but does not delete the current contents of memory. The program line numbers on disk are merged with the line numbers in memory. If two lines have the same number, only the line from the disk program is saved. After a MERGE command, the "merged" program resides in memory, and BASIC returns to command level.
KILL "filename"	Deletes the file from the disk. "filename" may be a program file, or a sequential or random access data file.
NAME	To change the name of a disk file, execute the NAME statement, NAME "oldfile" AS "newfile". NAME may be used with program files, random files, or sequential files.

B.2 PROTECTED FILES

If you wish to save a program in an encoded binary format, use the "Protect" option with the SAVE command. For example:

```
SAVE "MYPROG",P
```

A program saved this way cannot be listed or edited.

B.3 DISK DATA FILES = SEQUENTIAL AND RANDOM I/O

There are two types of disk data files that may be created and accessed by an ANSI BASIC program: sequential files and random access files.

B.3.1 Sequential Files

Sequential files are easier to create than random files but are limited in flexibility and speed when it comes to accessing the data. The data that is written to a sequential file is stored, one item after another (sequentially), in the order it is sent and is read back in the same way.

The statements and functions that are used with sequential files are:

OPEN	PRINT#	INPUT#	WRITE#
	PRINT# USING	LINE INPUT#	
CLOSE	EOF	LOC	

The following program steps are required to create a sequential file and access the data in the file:

- | | | |
|----|--|--------------------------------|
| 1. | OPEN the file in "O" mode. | OPEN "O",#1,"DATA" |
| 2. | Write data to the file
using the PRINT# statement.
(WRITE# may be used instead.) | PRINT#1,AS;BS;CS |
| 3. | To access the data in the
file, you must CLOSE the file
and reOPEN it in "I" mode. | CLOSE #1
OPEN "I",#1,"DATA" |
| 4. | Use the INPUT# statement to
read data from the sequential
file into the program. | INPUT#1,X\$,Y\$,Z\$ |

Program B-1 is a short program that creates a sequential file, "DATA", from information you input at the terminal.

```
10 OPEN "O",#1,"DATA"
20 INPUT "NAME";NS
25 IF NS="DONE" THEN END
30 INPUT "DEPARTMENT";DS
40 INPUT "DATE HIRED";HS
50 PRINT#1,NS;",";DS;",";HS
60 PRINT:GOTO 20
RUN
NAME? MICKEY MOUSE
DEPARTMENT? AUDIO/VISUAL AIDS
DATE HIRED? 01/12/72

NAME? SHERLOCK HOLMES
DEPARTMENT? RESEARCH
DATE HIRED? 12/03/65

NAME? EBENEZER SCROOGE
DEPARTMENT? ACCOUNTING
DATE HIRED? 04/27/78
NAME? SUPER MANN
DEPARTMENT? MAINTENANCE
DATE HIRED? 08/16/78
NAME? etc.

PROGRAM B-1 - CREATE A SEQUENTIAL DATA FILE
```

Now look at Program B-2. It accesses the file "DATA" that was created in Program B-1 and displays the name of everyone hired in 1978.

```

10 OPEN "I",#1,"DATA"
20 INPUT#1,NS,D$,HS
30 IF RIGHTS(H$,2)="78" THEN PRINT NS
40 GOTO 20
RUN
EBENEZER SCROOGE
SUPER MANN
Input past end in 20
Ok

```

PROGRAM B-2 - ACCESSING A SEQUENTIAL FILE

Program B-2 reads, sequentially, every item in the file. When all the data has been read, line 20 causes an "Input past end" error. To avoid getting this error, insert line 15 which uses the EOF function to test for end-of-file:

```
15 IF ECF(1) THEN END
```

and change line 40 to GOTO 15.

A program that creates a sequential file can also write formatted data to the disk with the PRINT# USING statement. For example, the statement

```
PRINT#1,USING"####.##",;A,B,C,D
```

could be used to write numeric data to disk without explicit delimiters. The comma at the end of the format string serves to separate the items in the disk file.

The LOC function, when used with a sequential file, returns the number of sectors that have been written to or read from the file since it was OPENed. A sector is a 124-byte block of data.

B.3.1.1 Adding Data To A Sequential File -

If you have a sequential file residing on disk and later want to add more data to the end of it, you cannot simply open the file in "O" mode and start writing data. As soon as you open a sequential file in "O" mode, you destroy its current contents. The following procedure can be used to add data to an existing file called "NAMES".

1. OPEN "NAMES" in "I" mode.
2. OPEN a second file called "COPY" in "O" mode.
3. Read in the data in "NAMES" and write it to "COPY".
4. CLOSE "NAMES" and KILL it.
5. Write the new information to "COPY".
6. Rename "COPY" as "NAMES" and CLOSE.
7. Now there is a file on disk called "NAMES" that includes all the previous data plus the new data you just added.

Program B-3 illustrates this technique. It can be used to create or add onto a file called NAMES. This program also illustrates the use of LINE INPUT# to read strings with embedded commas from the disk file. Remember, LINE INPUT# will read in characters from the disk until it sees a carriage return (it does not stop at quotes or commas) or until it has read 255 characters.

```

10 ON ERROR GOTO 2000
20 OPEN "I",#1,"NAMES"
30 REM IF FILE EXISTS, WRITE IT TO "COPY"
40 OPEN "O",#2,"COPY"
50 IF ECF(1) THEN 90
60 LINE INPUT#1,AS
70 PRINT#2,AS
80 GOTO 50
90 CLOSE #1
100 KILL "NAMES"
110 REM ADD NEW ENTRIES TO FILE
120 INPUT "NAME";NS
130 IF NS="" THEN 200 'CARRIAGE RETURN EXITS INPUT LOOP
140 LINE INPUT "ADDRESS? ";AS
150 LINE INPUT "BIRTHDAY? ";BS
160 PRINT#2,NS
170 PRINT#2,AS
180 PRINT#2,BS
190 PRINT:GOTO 120
200 CLOSE
205 REM CHANGE FILENAME BACK TO "NAMES"
210 NAME "COPY" AS "NAMES"
2000 IF ERR=53 AND ERL=20 THEN OPEN "O",#2,"COPY":RESUME 120
2010 ON ERROR GOTO 0

```

PROGRAM B-3 - ADDING DATA TO A SEQUENTIAL FILE

The error trapping routine in line 2000 traps a "File does not exist" error in line 20. If this happens, the statements that copy the file are skipped, and "COPY" is created as if it were a new file.

B.3.2 Random Files

Creating and accessing random files requires more program steps than sequential files, but there are advantages to using random files. One advantage is that random files require less room on the disk, because BASIC stores them in a packed binary format. (A sequential file is stored as a series of ASCII characters.)

The biggest advantage to random files is that data can be accessed randomly, i.e., anywhere on the disk -- it is not necessary to read through all the information, as with sequential files. This is possible because the information is stored and accessed in distinct units called records and each record is numbered.

The statements and functions that are used with random files are:

OPEN	FIELD	LSET/RSET	GET
PUT	CLOSE	LOC	
MKIS\$	CVI		
MKSS\$	CVS		
MKD\$	CVD		

B.3.2.1 Creating A Random File -

The following program steps are required to create a random file.

1. OPEN the file for random access ("R" mode). This example specifies a record length of 32 bytes. If the record length is omitted, the default is 128 bytes.
OPEN "R",#1,"FILE",32
 2. Use the FIELD statement to allocate space in the random buffer for the variables that will be written to the random file.
FIELD #1 20 AS NS,
4 AS AS\$, 8 AS PS
 3. Use LSET to move the data into the random buffer. Numeric values must be made into strings when placed in the buffer. To do this, use the "make" functions: MKI\$ to make an integer value into a string, MKSS\$ for a single precision value, and MKDS\$ for a double precision value.
LSET NS=X\$
LSET A\$=MKSS\$(AMT)
LSET PS=TELS
 4. Write the data from the buffer to the disk using the PUT statement.
PUT #1,CODE%

Look at Program B-4. It takes information that is input at the terminal and writes it to a random file. Each time the PUT statement is executed, a record is written to the file. The two-digit code that is input in line 30 becomes the record number.

NOTE

Do not use a FIELDed string variable in an INPUT or LET statement. This causes the pointer for that variable to point into string space instead of the random file buffer.

```
10 OPEN "R">#1,"FILE"
20 FIELD #1,20 AS NS, 4 AS AS$, 8 AS PS
30 INPUT "2-DIGIT CODE";CODE%
40 INPUT "NAME";XS
50 INPUT "AMOUNT";AMT
60 INPUT "PHONE";TELS:PRINT
70 LSET NS=X$
80 LSET A$=MK$(AMT)
90 LSET P$=TELS
100 PUT #1,CODE%
110 GOTO 30
```

PROGRAM B-4 - CREATE A RANDOM FILE

B.3.2.2 Access A Random File -

The following program steps are required to access a random file:

1. OPEN the file in "R" mode. OPEN "R",#1,"FILE",32
 2. Use the FIELD statement to allocate space in the random buffer for the variables that will be read from the file. FIELD #1 20 AS NS,
4 AS AS\$, 8 AS PS

NOTE:

In a program that performs both input and output on the same random file, you can often use just one OPEN statement and one FIELD statement.

3. Use the GET statement to move GET #1, CODE%
 the desired record into the
 random buffer.
4. The data in the buffer may PRINT N\$
 now be accessed by the program. PRINT CVS(A\$)
 Numeric values must be converted
 back to numbers using the
 "convert" functions: CVI for
 integers, CVS for single
 precision values, and CVD
 for double precision values.

Program B-5 accesses the random file "FILE" that was created in Program B-4. By inputting the three-digit code at the terminal, the information associated with that code is read from the file and displayed.

```

10 OPEN "R",#1,"FILE"
20 FIELD #1, 20 AS NS, 4 AS A$, 8 AS PS
30 INPUT "2-DIGIT CODE";CODE%
40 GET #1, CODE%
50 PRINT NS
60 PRINT USING "$$##.##";CVS(A$)
70 PRINT PS:PRINT
80 GOTO 30

```

PROGRAM B-5 - ACCESS A RANDOM FILE

The LOC function, with random files, returns the "current record number." The current record number is the last record number that was used in a GET or PUT statement. For example, the statement

```
IF LOC(1)>50 THEN END
```

ends program execution if the current record number in file#1 is higher than 50.

Program B-6 is an inventory program that illustrates random file access. In this program, the record number is used as the part number, and it is assumed the inventory will contain no more than 100 different part numbers. Lines 900-960 initialize the data file by writing CHR\$(255) as the first character of each record. This is used later (line 270 and line 500) to determine whether an entry already exists for that part number.

Lines 130-220 display the different inventory functions that the program performs. When you type in the desired function number, line 230 branches to the appropriate subroutine.

PROGRAM B-6 - INVENTORY

```

120 OPEN "R", #1, "INVEN.DAT", 39
125 FIELD #1, 1 AS FS, 30 AS DS, 2 AS QS, 2 AS RS, 4 AS PS
130 PRINT:PRINT "FUNCTIONS":PRINT
135 PRINT 1, "INITIALIZE FILE"
140 PRINT 2, "CREATE A NEW ENTRY"
150 PRINT 3, "DISPLAY INVENTORY FOR ONE PART"
160 PRINT 4, "ADD TO STOCK"
170 PRINT 5, "SUBTRACT FROM STOCK"
180 PRINT 6, "DISPLAY ALL ITEMS BELOW REORDER LEVEL"
190 PRINT:PRINT:INPUT "FUNCTION"; FUNCTION
225 IF (FUNCTION<1) OR (FUNCTION>6) THEN PRINT "BAD FUNCTION NUMBER":GOTO 130
230 ON FUNCTION GOSUB 900, 250, 390, 480, 560, 680
240 GOTO 220
250 REM BUILD NEW ENTRY
260 GOSUB 840

270 IF ASC(F$)<>255 THEN INPUT "OVERWRITE"; A$: IF A$ <> "Y" THEN RETURN
280 LSET FS=CHR$(0)
290 INPUT "DESCRIPTION"; DESC$
300 LSET DS=DESC$
310 INPUT "QUANTITY IN STOCK"; Q%
320 LSET QS=MKIS(Q%)
330 INPUT "REORDER LEVEL"; R%
340 LSET RS=MKIS(R%)
350 INPUT "UNIT PRICE"; P
360 LSET PS=MKSS(P)
370 PUT#1, PART%
380 RETURN
390 REM DISPLAY ENTRY
400 GOSUB 840
410 IF ASC(F$)=255 THEN PRINT "NULL ENTRY":RETURN
420 PRINT USING "PART NUMBER ####"; PART%
430 PRINT DS
440 PRINT USING "QUANTITY ON HAND #####"; CVI(Q$)
450 PRINT USING "REORDER LEVEL #####"; CVI(R$)
460 PRINT USING "UNIT PRICE $##.##"; CVS(PS)
470 RETURN
480 REM ADD TO STOCK
490 GOSUB 840
500 IF ASC(F$)=255 THEN PRINT "NULL ENTRY":RETURN
510 PRINT DS: INPUT "QUANTITY TO ADD "; A%
520 Q%=CVI(Q$)+A%
530 LSET QS=MKIS(Q%)
540 PUT#1, PART%
550 RETURN
560 REM REMOVE FROM STOCK
570 GOSUB 840
580 IF ASC(F$)=255 THEN PRINT "NULL ENTRY":RETURN
590 PRINT DS
600 INPUT "QUANTITY TO SUBTRACT"; S%
610 Q%=CVI(Q$)
620 IF (Q%-S%)<0 THEN PRINT "ONLY"; Q%; " IN STOCK":GOTO 600
630 Q%=Q%-S%
640 IF Q%=<CVI(R$) THEN PRINT "QUANTITY NOW"; Q%; " REORDER LEVEL"; CVI(R$)
650 LSET QS=MKIS(Q%)
660 PUT#1, PART%

```

```
670 RETURN
680 DISPLAY ITEMS BELOW REORDER LEVEL
690 FCR I=1 TC 100
710 GET#1,I
720 IF CVI(Q$)<CVI(R$) THEN PRINT D$;" QUANTITY";CVI(Q$) TAB(50) "REORDER LEV"
730 NEXT I
740 RETURN
840 INPUT "PART NUMBER";PART%
850 IF(PART%<1)OR(PART%>100) THEN PRINT "BAD PART NUMBER":GOTO 840 ELSE GET#1
890 END
900 REM INITIALIZE FILE
910 INPUT "ARE YOU SURE";B$:IF B$<>"Y" THEN RETURN
920 LSET FS=CHR$(255)
930 FOR I=1 TO 100
940 PUT#1,I
950 NEXT I
960 RETURN
```

APPENDIX C

Assembly Language Subroutines

All versions of ANSI BASIC have provisions for interfacing with assembly language subroutines. The USR function allows assembly language subroutines to be called in the same way BASIC's intrinsic functions are called.

C.1 MEMORY ALLOCATION

Memory space must be set aside for an assembly language subroutine before it can be loaded. During initialization, enter the highest memory location minus the amount of memory needed for the assembly language subroutine(s). BASIC uses all memory available from its starting location up, so only the topmost locations in memory can be set aside for user subroutines.

When an assembly language subroutine is called, the stack pointer is set up for 8 levels (16 bytes) of stack storage. If more stack space is needed, BASIC's stack can be saved and a new stack set up for use by the assembly language subroutine. BASIC's stack must be restored, however, before returning from the subroutine.

The assembly language subroutine may be loaded into memory by means of the system monitor or the BASIC POKE statement. Through FLP-80DOS use ASM to assemble the routine, and LINK to create the binary file, then use the GET command to load the routine at a given address.

C.2 USR FUNCTION CALLS

In MOSTEK ANSI BASIC , the format of the USR function is

USR[<digit>](argument)

where DIGIT> is from 0 to 9 and the argument is any numeric or string expression. <digit> specifies which USR routine is being called, and corresponds with the digit supplied in the DEF USR statement for that routine. If <digit> is omitted, USR0 is assumed. The address given in the DEF USR statement determines the starting address of the subroutine.

When the USR function call is made, register A contains a value that specifies the type of argument that was given. The value in A may be one of the following:

<u>Value in A</u>	<u>Type of Argument</u>
2	Two-byte integer (two's complement)
3	String
4	Single precision floating point number
8	Double precision floating point number

If the argument is a number, the [H,L] register pair points to the Floating Point Accumulator (FAC) where the argument is stored.

If the argument is an integer:

FAC-3 contains the lower 8 bits of the argument and
FAC-2 contains the upper 8 bits of the argument.

If the argument is a single precision floating point number:

FAC-3 contains the lowest 8 bits of mantissa and

FAC-2 contains the middle 8 bits of mantissa and FAC-1 contains the highest 7 bits of mantissa with leading 1 suppressed (implied). Bit 7 is the sign of the number (0=positive, 1=negative). FAC is the exponent minus 128, and the binary point is to the left of the most significant bit of the mantissa.

If the argument is a double precision floating point number:

FAC-7 through FAC-4 contain four more bytes of mantissa (FAC-7 contains the lowest 8 bits).

If the argument is a string, the [D,E] register pair points to 3 bytes called the "string descriptor." Byte 0 of the string descriptor contains the length of the string (0 to 255). Bytes 1 and 2, respectively, are the lower and upper 8 bits of the string starting address in string space.

CAUTION: If the argument is a string literal in the program, the string descriptor will point to program text. Be careful not to alter or destroy your program this way. To avoid unpredictable results, add +"" to the string literal in the program. Example:

```
A$ = "ANSI BASIC"+""
```

This will copy the string literal into string space and will prevent alteration of program text during a subroutine call.

Usually, the value returned by a USR function is the same type (integer, string, single precision or double precision) as the argument that was passed to it. However, calling the MAKINT routine returns the integer in [H,L] as the value of the function, forcing the value returned by the function to be integer. To execute MAKINT, use the following sequence to return from the subroutine:

MAKINT	EQU	2DB8H	;MAKINT address in BASIC 5.3
	PUSH	HL	;save value to be returned
	LD	HL,MAKINT	;get address of MAKINT routine
	EX	(SP),HL	;save return on stack and
	RET		;return

Also, the argument of the function, regardless of its type, may be forced to an integer by calling the FRCINT routine to get the integer value of the argument in [H,L]. Execute the following routine:

FRCINT	EQU	2D56H	;FRCINT address in BASIC 5.3
	LD	HL,SUB	;get address of subroutine
	PUSH	HL	;place on stack
	LD	HL,FRCINT	;get address of FRCINT
	JP	(HL)	
SUB:	.	.	.

C.3 CALL STATEMENT

ANSI BASIC user function calls may also be made with the CALL statement. The calling sequence used is the same as that in Microsoft's FORTRAN, COBOL and BASIC compilers.

A CALL statement with no arguments generates a simple "CALL" instruction. The corresponding subroutine should return via a simple "RET." (CALL and RET are Z80 instructions, see a Z80 reference manual for details.)

A subroutine CALL with arguments results in a somewhat more complex calling sequence. For each argument in the CALL argument list, a parameter is passed to the subroutine.

That parameter is the address of the low byte of the argument. Therefore, parameters always occupy two bytes each, regardless of type.

The method of passing the parameters depends upon the number of parameters to pass:

1. If the number of parameters is less than or equal to 3, they are passed in the registers. Parameter 1 will be in HL, 2 in DE (if present), and 3 in BC (if present).
2. If the number of parameters is greater than 3, they are passed as follows:
 1. Parameter 1 in HL.
 2. Parameter 2 in DE.
 3. Parameters 3 through n in a contiguous data block. BC will point to the low byte of this data block (i.e., to the low byte of parameter 3).

Note that, with this scheme, the subroutine must know how many parameters to expect in order to find them. Conversely, the calling program is responsible for passing the correct number of parameters. There are no checks for the correct number or type of parameters.

If the subroutine expects more than 3 parameters, and needs to transfer them to a local data area, there is a system subroutine which will perform this transfer. This argument transfer routine is named \$AT (located in the FORTRAN library, FORLIB.REL), and is called with HL pointing to the local data area, BC pointing to the third parameter, and A containing the number of arguments to transfer (i.e., the total number of arguments minus 2). The subroutine is

responsible for saving the first two parameters before calling AT. For example, if a subroutine expects 5 parameters, it should look like:

```

SUBR: LD      (P1),HL      ;SAVE PARAMETER 1
      EX      DE,HL
      LD      (P2),HL      ;SAVE PARAMETER 2
      LD      A,3          ;NO. OF PARAMETERS LEFT
      LD      HL,P3        ;pointer to local area
      CALL   AT           ;TRANSFER THE OTHER 3 PARAMETERS
      .
      .
      .
      [Body of subroutine]
      .
      .
      .
      .
      RET      ;RETURN TO CALLER
P1:  DEFS   2          ;SPACE FOR PARAMETER 1
P2:  DEFS   2          ;SPACE FOR PARAMETER 2
P3:  DEFS   6          ;SPACE FOR PARAMETERS 3-5

```

A listing of the argument transfer routine AT follows.

```

00100  ;      ARGUMENT TRANSFER
00200  ;[B,C]  POINTS TO 3RD PARAM.
00300  ;[H,L]  POINTS TO LOCAL STORAGE FOR PARAM 3
00400  ;[A]    CONTAINS THE # OF PARAMS TO XFER (TOTAL-2)
00500
00600
00700  ENTRY   AT
00800  AT:    EX      DE,HL      ;SAVE [H,L] IN [D,E]
00900  LD      H,B
01000  LD      L,C          ;[H,L] = PTR TO PARAMS
01100  AT1:   LD      C,(HL)
01200  INC    HL
01300  LD      B,(HL)
01400  INC    HL          ;[B,C] = PARAM ADR
01500  EX      DE,HL      ;[H,L] POINTS TO LOCAL STORAGE
01600  LD      (HL),C
01700  INC    HL
01800  LD      (HL),B
01900  INC    HL          ;STORE PARAM IN LOCAL AREA
02000  LD      DE,HL      ;SINCE GOING BACK TO AT1
02100  DEC    A           ;TRANSFERRED ALL PARAMS?
02200  JR     NZ,AT1      ;NO, COPY MORE
02300  RET

```

When accessing parameters in a subroutine, don't forget that they are pointers to the actual arguments passed.

NOTE

It is entirely up to the programmer to see to it that the arguments in the calling program match in number, type, and length with the parameters expected by the subroutine. This applies to BASIC subroutines, as well as those written in assembly language.

C.4 INTERRUPTS

Assembly language subroutines can be written to handle interrupts. All interrupt handling routines should save the stack, register A-L and the PSW. Interrupts should always be re-enabled before returning from the subroutine., since an interrupt automatically disables all further interrupts once it is received. The user should be aware of which interrupt vectors are free in the particular version of BASIC that has been supplied.

APPENDIX D

Converting Programs to ANSI BASIC

If you have programs written in a BASIC other than ANSI BASIC, some minor adjustments may be necessary before running them with ANSI BASIC. Here are some specific things to look for when converting BASIC programs.

D.1 STRING DIMENSIONS

Delete all statements that are used to declare the length of strings. A statement such as `DIM A$(I,J)`, which dimensions a string array for J elements of length I, should be converted to the ANSI BASIC statement `DIM A$(J)`.

Some BASICs use a comma or ampersand for string concatenation. Each of these must be changed to a plus sign, which is the operator for ANSI BASIC string concatenation.

In ANSI BASIC, the `MIDS`, `RIGHT$`, and `LEFT$` functions are used to take substrings of strings. Forms such as `A$(I)` to access the Ith character in `A$`, or `A$(I,J)` to take a substring of `A$` from position I to position J, must be changed as follows:

<u>Other BASIC</u>	<u>ANSI BASIC</u>
<code>X\$=A\$(I)</code>	<code>X\$=MIDS(A\$,I,1)</code>
<code>X\$=A\$(I,J)</code>	<code>X\$=MIDS(A\$,I,J-I+1)</code>

If the substring reference is on the left side of an assignment and `X$` is used to replace characters in `A$`, convert as follows:

<u>Other BASIC</u>	<u>ANSI BASIC</u>
<code>A\$(I)=X\$</code>	<code>MIDS(A\$,1,1)=X\$</code>
<code>A\$(I,J)=X\$</code>	<code>MIDS(A\$,I,J-I+1)=X\$</code>

D.2 MULTIPLE ASSIGNMENTS

Some BASICs allow statements of the form:

```
10 LET B=C=0
```

to set B and C equal to zero. ANSI BASIC would interpret the second equal sign as a logical operator and set B equal to -1 if C equaled 0.

Instead, convert this statement to two assignment statements:

```
10 C=0:B=0
```

D.3 MULTIPLE STATEMENTS

Some BASICs use a backslash (\) to separate multiple statements on a line. With ANSI BASIC, be sure all statements on a line are separated by a colon (:).

D.4 MAT FUNCTIONS

Programs using the MAT functions available in some BASICs must be rewritten using FOR...NEXT loops to execute properly.

APPENDIX E

Summary of Error Codes and Error Messages

<u>Code</u>	<u>Number</u>	<u>Message</u>
BS	9	Subscript out of range An array element is referenced either with a subscript that is outside the dimensions of the array, or with the wrong number of subscripts.
CN	17	Can't continue An attempt is made to continue a program that: 1. has halted due to an error, 2. has been modified during a break in execution, or 3. does not exist.
DD	10	Redimensioned array Two DIM statements are given for the same array, or a DIM statement is given for an array after the default dimension of 10 has been established for that array.
FC	5	Illegal function call A parameter that is out of range is passed to a math or string function. An FC error may also occur as the result of: 1. a negative or unreasonably large subscript 2. a negative or zero argument with LOG 3. a negative argument to SQR 4. a negative mantissa with a non-integer exponent 5. a call to a USR function for which the starting address has not yet been given 6. an improper argument to MIDS, LEFT\$, RIGHTS\$, INP, OUT, WAIT, PEEK, POKE, TAB, SPC, STRINGS\$, SPACES\$, INSTR, or ON...GOTO.

ID	12	Illegal direct
		A statement that is illegal in direct mode is entered as a direct mode command.
NF	1	NEXT without FOR A variable in a NEXT statement does not correspond to any previously executed, unmatched FOR statement variable.
OD	4	Out of data A READ statement is executed when there are no DATA statements with unread data remaining in the program.
OM	7	Out of memory A program is too large, has too many FOR loops or GOSUBs, too many variables, or expressions that are too complicated.
OS	14	Out of string space String variables exceed the allocated amount of string space. Use CLEAR to allocate more string space, or decrease the size and number of strings.
OV	6	Overflow The result of a calculation is too large to be represented in ANSI BASIC's number format. If underflow occurs, the result is zero and execution continues without an error.
SN	2	Syntax error A line is encountered that contains some incorrect sequence of characters (such as unmatched parenthesis, misspelled command or statement, incorrect punctuation, etc.).
ST	16	String formula too complex A string expression is too long or too complex. The expression should be broken into smaller expressions.
TM	13	Type mismatch A string variable name is assigned a numeric value or vice versa; a function that expects a numeric argument is given a string argument or vice versa.

- RG 3 Return without GOSUB
A RETURN statement is encountered for which there is no previous, unmatched GOSUB statement.
- UF 18 Undefined user function
A USR function is called before the function definition (DEF statement) is given.
- UL 8 Undefined line
A line reference in a GOTO, GOSUB, IF...THEN...ELSE or DELETE is to a nonexistent line.
- /0 11 Division by zero
A division by zero is encountered in an expression, or the operation of involution results in zero being raised to a negative power. Machine infinity with the sign of the numerator is supplied as the result of the division, or positive machine infinity is supplied as the result of the involution, and execution continues.
- 19 No RESUME
An error trapping routine is entered but contains no RESUME statement.
- 20 RESUME without error
A RESUME statement is encountered before an error trapping routine is entered.
- 21 Unprintable error
An error message is not available for the error condition which exists. This is usually caused by an ERROR with an undefined error code.
- 22 Missing operand
An expression contains an operator with no operand following it.
- 23 Line buffer overflow
An attempt is made to input a line that has too many characters.
- 26 FOR without NEXT
A FOR was encountered without a matching NEXT.
- 29 WHILE without WEND
A WHILE statement does not have a matching WEND.

30 WEND without WHILE
A WEND was encountered without a matching WHILE.

Disk Errors

- 50 Field overflow
A FIELD statement is attempting to allocate more bytes than were specified for the record length of a random file.
- 51 Internal error
An internal malfunction has occurred in Disk ANSI BASIC. Report to Mostek the conditions under which the message appeared.
- 52 Bad file number
A statement or command references a file with a file number that is not OPEN or is out of the range of file numbers specified at initialization.
- 53 File not found
A LOAD, KILL or OPEN statement references a file that does not exist on the current disk.
- 54 Bad file mode
An attempt is made to use PUT, GET, or LOF with a sequential file, to LOAD a random file or to execute an OPEN with a file mode other than I, O, or R.
- 55 File already open
A sequential output mode OPEN is issued for a file that is already open; or a KILL is given for a file that is open.
- 57 Disk I/O error
An I/O error occurred on a disk I/O operation. It is a fatal error, i.e., the operating system cannot recover from the error.
- 58 File already exists
The filename specified in a NAME statement is identical to a filename already in use on the disk.
- 61 Disk full
All disk storage space is in use.

- 62 Input past end
An INPUT statement is executed after all the data in the file has been INPUT, or for a null (empty) file. To avoid this error, use the EOF function to detect the end of file.
- 63 Bad record number
In a PUT or GET statement, the record number is either greater than the maximum allowed (32767) or equal to zero.
- 64 Bad file name
An illegal form is used for the filename with LOAD, SAVE, KILL, or OPEN (e.g., a filename with too many characters).
- 66 Direct statement in file
A direct statement is encountered while LOADING an ASCII-format file. The LOAD is terminated.
- 67 Too many files
An attempt is made to create a new file (using SAVE or OPEN) when all 255 directory entries are full.



APPENDIX F

Mathematical Functions

Derived Functions

Functions that are not intrinsic to ANSI BASIC may be calculated as follows.

<u>Function</u>	<u>ANSI BASIC Equivalent</u>
SECANT	$\text{SEC}(X)=1/\text{COS}(X)$
COSECANT	$\text{CSC}(X)=1/\text{SIN}(X)$
COTANGENT	$\text{COT}(X)=1/\text{TAN}(X)$
INVERSE SINE	$\text{ARCSIN}(X)=\text{ATN}(X/\text{SQR}(-X*X+1))$
INVERSE COSINE	$\text{ARCCOS}(X)=-\text{ATN}(X/\text{SQR}(-X*X+1))+1.5708$
INVERSE SECANT	$\text{ARCSEC}(X)=\text{ATN}(X/\text{SQR}(X*X-1)) + \text{SGN}(\text{SGN}(X)-1)*1.5708$
INVERSE COSECANT	$\text{ARCCSC}(X)=\text{ATN}(X/\text{SQR}(X*X-1)) + (\text{SGN}(X)-1)*1.5708$
INVERSE CCTANGENT	$\text{ARCCOT}(X)=\text{ATN}(X)+1.5708$
HYPERBOLIC SINE	$\text{SINH}(X)=(\text{EXP}(X)-\text{EXP}(-X))/2$
HYPERBOLIC COSINE	$\text{COSH}(X)=(\text{EXP}(X)+\text{EXP}(-X))/2$
HYPERBOLIC TANGENT	$\text{TANH}(X)=\text{EXP}(-X)/(\text{EXP}(X)+\text{EXP}(-X))*2+1$
HYPERBOLIC SECANT	$\text{SECH}(X)=2/(\text{EXP}(X)+\text{EXP}(-X))$
HYPERBOLIC COSECANT	$\text{CSCH}(X)=2/(\text{EXP}(X)-\text{EXP}(-X))$
HYPERBOLIC COTANGENT	$\text{COTh}(X)=\text{EXP}(-X)/(\text{EXP}(X)-\text{EXP}(-X))*2+1$
INVERSE HYPERBOLIC SINE	$\text{ARCSINH}(X)=\text{LOG}(X+\text{SQR}(X*X+1))$
INVERSE HYPERBOLIC COSINE	$\text{ARCCOSH}(X)=\text{LOG}(X+\text{SQR}(X*X-1))$
INVERSE HYPERBOLIC TANGENT	$\text{ARCTANH}(X)=\text{LOG}((1+X)/(1-X))/2$
INVERSE HYPERBOLIC SECANT	$\text{ARCSECH}(X)=\text{LOG}((\text{SQR}(-X*X+1)+1)/X)$
INVERSE HYPERBOLIC COSECANT	$\text{ARCCSCH}(X)=\text{LOG}((\text{SGN}(X)*\text{SQR}(X*X+1)+1)/X)$
INVERSE HYPERBOLIC COTANGENT	$\text{ARCCOTH}(X)=\text{LOG}((X+1)/(X-1))/2$

APPENDIX G

ASCII Character Codes

ASCII Code	Character	ASCII Code	Character	ASCII Code	Character
000	NUL	043	+	086	V
001	SOH	044	,	087	W
002	STX	045	-	088	X
003	ETX	046	.	089	Y
004	EOT	047	/	090	Z
005	ENQ	048	0	091	[
006	ACK	049	1	092	\
007	BEL	050	2	093]
008	BS	051	3	094	^
009	HT	052	4	095	-
010	LF	053	5	096	a
011	VT	054	6	097	b
012	FF	055	7	098	c
013	CR	056	8	099	d
014	SO	057	9	100	e
015	SI	058	:	101	f
016	DLE	059	;	102	g
017	DC1	060	<	103	h
018	DC2	061	=	104	i
019	DC3	062	>	105	j
020	DC4	063	?	106	k
021	NAK	064	�	107	l
022	SYN	065	A	108	m
023	ETB	066	B	109	n
024	CAN	067	C	110	o
025	EM	068	D	111	p
026	SUB	069	E	112	q
027	ESCAPE	070	F	113	r
028	FS	071	G	114	s
029	GS	072	H	115	t
030	RS	073	I	116	u
031	US	074	J	117	v
032	SPACE	075	K	118	w
033	!	076	L	119	x
034	"	077	M	120	y
035	#	078	N	121	z
036	\$	079	O	122	�
037	%	080	P	123	�
038	&	081	Q	124	�
039	'	082	R	125	�
040	(083	S	126	~
041)	084	T	127	DEL
042	*	085	U		

ASCII codes are in decimal.

LF=Line Feed, FF=Form Feed, CR=Carriage Return, DEL=Rubout

INDEX

ABS	3-2
Addition	1-9
ALL	2-4, 2-8
Arctangent	3-3
Array variables	1-7, 2-8, 2-16
Arrays	1-7, 2-22
ASC	3-2
ASCII codes	3-2, 3-4, G-1
ASCII format	2-48, 2-77
Assembly language subroutines	2-3, 2-14, 2-47, 3-21, 3-23, C-1
ATN	3-3
AUTO	1-2, 2-2
Boolean operators	1-12
CALL	2-3, C-4
Carriage return	1-3, 2-35, 2-40 to 2-41, 2-84 to 2-86
CDBL	3-3
CHAIN	2-4, 2-8
Character set	1-3
CHR\$	3-4
CINT	3-4
CLEAR	2-6, A-1
CLOSE	2-7, B-3, B-8
Command level	1-1
COMMON	2-4, 2-8
Concatenation	1-14
Constants	1-4
CONT	2-9, 2-40
Control characters	1-4
Control-A	2-20
COS	3-5
CSNG	3-5
CVD	3-6, B-8
CVI	3-6, B-8
CVS	3-6, B-8
DATA	2-10, 2-74

DEF FN	2-11
DEF USR	2-14, 3-21
DEFDBL	1-7, 2-13
DEFINT	1-7, 2-13
DEFSNG	1-7, 2-13
DEFSTR	1-7, 2-13
DEINT	C-1
DELETE	1-2, 2-4, 2-15
DIM	2-16
Direct mode	1-1, 2-33, 2-53
Division	1-10
Double precision	1-3, 2-15, 2-59, 3-3, A-1
DSKF	3-6
EDIT	1-2, 2-17
Edit mode	1-4, 2-17
END	2-7, 2-10, 2-21, 2-31
EOF	3-7, B-3, B-5
ERASE	2-22
ERL	2-23
ERR	2-23
ERROR	2-24
Error codes	1-15, 2-24 to 2-25, E-1
Error messages	1-15, E-1
Error trapping	2-24 to 2-25, 2-53, 2-75, B-7
Escape	1-3, 2-18
EXP	3-7
Exponentiation	1-10 to 1-11, 3-7
Expressions	1-9
FIELD	2-26, B-8
FILES	2-27
FIX	3-7
FOR...NEXT	2-29, A-1
FRCINT	C-1, C-4, D-4
FRE	3-8
Functions	1-14, 2-13, 3-1, F-1
GET	2-28, 2-31, A-3, B-8,
GIVABF	C-1 to C-2
GOSUB	2-31
GOTO	2-32 to 2-33
HEX\$	3-8
Hexadecimal	1-5, 3-8
IF...GOTO	2-33
IF...THEN	2-23, 2-33
IF...THEN...ELSE	2-33
Indirect mode	1-1
INP	3-9
INPUT	2-9, 2-26, 2-35, A-2, B-9

INPUT\$	3-9
INPUT#	2-37, B-3
INSTR	3-10
INT	3-7, 3-10
Integer	3-4, 3-7, 3-10
Integer division	1-10
Interrupts	C-6
 KILL	2-38, B-2
 LEFT\$	3-11
LEN	3-11
LET	2-26, 2-39, B-9
Line feed	1-2, 2-35, 2-40 to 2-41, 2-85 To 2-86
LINE INPUT	2-40
LINE INPUT#	2-41, B-3
Line numbers	1-1 to 1-2, 2-2, 2-72
Line printer	2-44, 2-46, 2-84, 3-13, A-2
Lines	1-1
LIST	1-2, 2-42
LLIST	2-44
LOAD	2-45, 2-77, B-1
LOC	3-12, B-3, B-5, B-10
LOF	3-12
LOG	3-13
Logical operators	1-12
Loops	2-28, 2-82
LPOS	2-84, 3-13
LPRINT	2-46, 2-83
LPRINT USING	2-46
LSET	2-47, B-8
 MAKINT	C-1, C-3
MERGE	2-4, 2-48, B-2
MIDS	2-49, 3-13, D-1
MKDS	3-14, B-8
MKIS	3-14, B-8
MKSS\$	3-14, B-8
MOD operator	1-10
Modulus arithmetic	1-10
Multiplication	1-10
 NAME	2-50
Negation	1-10

NEW	2-7, 2-51
NULL	2-52
Numeric constants	1-4
Numeric variables	1-7
OCT\$	3-14
Octal	1-5, 3-14
ON ERROR GOTO	2-53
ON...GOSUB	2-54
ON...GOTO	2-54, A-1
OPEN	2-7, 2-27, 2-55, B-3, B-8
Operators	1-9, 1-11 to 1-14
OPTION BASE	2-56
OUT	2-57
Overflow	1-11, 3-7, 3-21, A-1
Overlay	2-4
Paper tape	2-52
PEEK	2-58, 3-15
POKE	2-58, 3-15
POS	2-84, 3-15
PRINT	2-59, A-1
PRINT USING	2-61, A-2
PRINT#	2.65, B-3
PRINT# USING	2.65, B-3, B-5
Protected files	2-77, A-2, B-2
PUT	2-26, 2-67, B-8
Random files	2-26, 2-30, 2-38, 2-47, 2-55, 2-67, 3-12, 3-14, B-7
Random numbers	2-68, 3-16
RANDOMIZE	2-68, 3-16, A-1
READ	2-69, 2-74
Relational operators	1-11
REM	2-71
RENUM	2-4, 2-23, 2-72
RESET	2-73
RESTORE	2-74
RESUME	2-75
RETURN	2-31
RIGHTS	3-16
RND	2-68, 3-16, A-1
RSET	2-47, B-8
Rubout	1-3, 1-15, 2-18
RUN	2-76 to 2-77, B-2
SAVE	2-45, 2-76 to 2-77, B-1
Sequential files	2-37 to 2-38, 2-41, 2-55, 2-65, 2-85, 3-6, 3-12, B-3

SGN	3-17
SIN	3-17
Single precision	1-5, 2-15, 2-59; 3-5, A-1
SPACES	3-18
SPC	3-18
SQR	3-19
STOP	2-9, 2-21, 2-31, 2-78
STR\$	3-19
String constants	1-4
String functions	3-6, 3-10 to 3-11, 3-13, 3-16, 3-19, 3-22
String operators	1-14
String space	2-6, 3-8, A-1, B-9
String variables	1-7, 2-13, 2-40 to 2-41
STRINGS	3-20
Subroutines	2-3, 2-31, 2-54, C-1
Subscripts	1-6, 2-16, 2-57
Subtraction	1-10
SWAP	2-79
SYSTEM	2-80
TAB	3-20
Tab	1-3 to 1-4
TAN	3-21
TROFF	2-81
TRON	2-81
USR	2-14, 3-21, C-1
USRLOC	C-2
VAL	3-22
Variables	1-6
VARPTR	3-23
WAIT	2-82
WEND	2-83
WHILE	2-83
WIDTH	2-84, A-2
WIDTH LPRINT	2-84, A-2
WRITE	2-85
WRITE#	2-86, B-3



MOSTEK®
Z80-F8 Covering the full
3870 spectrum of
microcomputer
applications.

1215 W. Crosby Rd. • Carrollton, Texas 75006 • 214/323-6000
In Europe, Contact: MOSTEK Brussels
150 Chaussee de la Hulpe, B1170, Belgium;
Telephone: 660.69.24

Mostek reserves the right to make changes in specifications at any time and without notice. The information furnished by Mostek in this publication is believed to be accurate and reliable. However, no responsibility is assumed by Mostek for its use; nor for any infringements of patents or other rights of third parties resulting from its use. No license is granted under any patents or patent rights of Mostek.

PRINTED IN USA March 1980
Publication No. MK79708
STD No. 8023-79708-0505

Copyright 1980 by Mostek Corporation
All rights Reserved